

ANSYS分析实例与工程应用

邢静忠 王永岗 陈晓霞 等编著



机 械 工 业 出 版 社

本书以命令行方式通过大量教学实例和工程应用实例,介绍了建立模型、求解和结果后处理的全过程,内容覆盖杆系、梁系、板壳和实体结构的静力分析和动力学分析内容。以实例讲解为主线,辅助以理论分析和计算结果比较评价,使读者在学习 ANSYS 命令行操作的同时,不断提高评价计算结果的能力和理论修养,全书算例注重问题的力学模型分析和结果分析。配套光盘给读者提供了练习求解问题的演练机会,同时读者可以不断改变问题的求解条件,分析评价计算结果的正确性和准确性,以达到举一反三的练习效果。工程实例可以帮助读者了解工程问题的模型简化和求解过程。

本书最大的特点是,用命令行方式结合实例讲解常用命令和求解过程,所选择的 64 个教学算例覆盖了常见工程结构分析领域。从知识层次上覆盖了从理论力学、材料力学、结构力学直到弹性力学范畴。较多的例题不仅可以帮助读者学习和熟练 ANSYS 软件的使用,还可以帮助读者理解有限元的基本思想,积累实际操作经验,不断提高分析处理问题的能力。书末的结构单元参考和常用命令速查也是结构分析必备的工具箱。

本书适合大学本科二年级以上的学生和研究生使用,也可以供从事结构分析和设计的其他人员参考。

图书在版编目(CIP)数据

中国版本图书馆 CIP 数据核字(2003)第 083125 号

机械工业出版社(北京市百万庄大街 22 号 邮政编码 100037)

策划编辑:吴宏伟 版式设计: 责任校对:

封面设计: 责任印刷:

北京机工印刷厂印刷·新华书店北京发行所发行

2004 年 1 月第 1 版第 1 次印刷

787mm×1092mm 1/16· 印张· 千字

0001-5000 册

定价: .00 元

凡购本图书,如有缺页、倒页、脱页,由本社发行部调换

本社购书热线电话:(010) 68993821、88379646

封面无防伪标均为盗版

前 言

随着计算机技术的日益普及，计算机工具在提高社会生产力方面发挥了越来越重要的作用，特别是CAD/CAE/CAM在工业界的日益成熟和普及，极大地提高了工业设计和生产效率，越来越多的人逐渐熟悉这些强有力的工具，同时对这些技术给工业界的推动作用逐渐认可。越来越多从事着从设计到生产各个环节的工程人员和在校学生逐步成为这一新技术的主要用户群体。

有限元分析技术已经发展成为计算机辅助分析CAE的核心。用CAE方法可以减少或避免物理测试过程，通过计算机模拟最恶劣载荷和工况下零件或结构的工作情况，准确地计算其应力应变，使产品在设计阶段就能够对其数学模型的各项性能进行评估，及早发现设计上存在的问题，从而大大缩短设计开发周期。

特别是采用有限元分析技术及其优化技术，能够改进结构设计参数，使其在满足强度和刚度的情况下具有最合理的结构。在应用于新产品开发和老产品改造方面，能够提供对其强度、应力分布状况的分析，利用优化设计方法对其进行形状和结构优化设计，从而在设计上提供技术支持和理论指导。

作为国内最为流行的ANSYS软件，在工程计算、教学实践和科学研究方面已经积累了大量的应用实例。特别是在校学生越来越普及了对ANSYS软件的了解和应用，不少学校购买了ANSYS软件，并开展了一系列学习和推广ANSYS应用的热潮，互联网上的ANSYS讨论园地也日益火爆。

在这样一个红红火火的ANSYS流行时期，本书可以让初学者快速、全面地掌握ANSYS的使用要领，完成最一般的工程结构分析。本书以ANSYS高级分析为主要内容，结合实例讲解ANSYS的操作命令和分析过程，是为初学者提供的、能够快速提高ANSYS分析计算能力的实用教程。

本书采用ANSYS命令行方式，从最经典的力学问题入手，采用理论分析和ANSYS计算相结合，重点分析模型简化，对计算结果进行分析和比较，使读者在对问题有一个大致的把握后，开始建模计算，并对计算结果的正确性和准确性作出准确判断。

学会ANSYS软件的使用并不难，得到和计算模型一致的计算结果也不太难，最困难的是：得到和实际模型一致的结果，并能分析出计算结果和实际情况的差异。解决这个困难的方法只有一条，就是大量分析算例积累起来的实际计算经验。本书的算例就是帮助读者在静力分析和动力分析方面，对杆、梁、板壳和实体结构模型及各类结构分析单元特性开始技术积累和丰富经验的过程。相信读者在读完本书，并结合所附光盘算例的演练后，一定会变成一位ANSYS分析高手。不仅在软件操作方面，更多的是在结果评价和模型理论分析方面成为一个经验丰富的专家。

本书在编写过程中得到了仿真互动网站上的许多有限元爱好者的帮助和支持。同时我也非常感谢兰州理工大学的同事和我的家人对本人工作的支持和理解。本书的部分资料来源于ANSYS验证算例和参考文献，特向他们表示感谢。

同时也感谢机械工业出版社北京时代金科科技有限公司对作者的信任和责任编辑为本

书出版所付出的辛勤工作！

全书由邢静忠、王永岗和陈晓霞主编，朱爱红、宋曦、黄达文、杨静宁、滕兆春、徐燕、张敏、马维俊、徐永君、白金泽、董永香、宋慧芳、杨丽敏等参与了部分编写工作。

鉴于时间仓促和作者认识的局限，书中错误在所难免，欢迎读者批评指正。

读者如需书中的源程序，请到以下网站下载：

<http://www.cmpbook.com/download/13746.zip>

或<http://hsingjzh.welan.com>

编者

2004年5月 北京

目 录

第1篇 教学实例篇

第1章 简单拉压杆结构	1
1.1 铰接杆在外力作用下的变形	2
1.2 人字形屋架的静力分析	6
1.3 超静定拉压杆的反力计算	10
1.4 平行杆件与刚性梁连接的热应力问题	13
1.5 端部有间隙的杆的热膨胀	17
第2章 梁的弯曲问题	21
2.1 等截面简单超静定梁的平面弯曲分析	21
2.2 工字形截面外伸梁的平面弯曲	29
2.3 矩形截面梁的纵横弯曲分析	35
2.4 悬臂梁的双向弯曲	45
2.5 圆形截面悬臂杆的弯扭组合变形	60
2.6 悬臂等强度梁的弯曲	64
2.7 弹性地基半无限长梁在端部力和力偶作用下的变形	74
2.8 偏心受压杆的大变形分析	78
第3章 杆系稳定性计算	85
3.1 利用梁单元计算压杆稳定性	85
3.2 利用实体单元计算压杆稳定性	89
3.3 悬臂压杆的过曲屈分析	93
3.4 平面钢架的平面外失稳	103
第4章 实体模型应力分析	115
4.1 均布荷载作用下深梁的变形和应力	115
4.2 一对集中力作用下的圆环	128
4.3 用实体单元分析变截面杆的拉伸	140
4.4 用二维实体单元分析等截面悬臂梁的平面弯曲	148
4.5 变截面悬臂梁在端部集中力作用下的平面静力分析	155
4.6 纯弯曲悬臂曲梁的二维静力分析	164
4.7 端部集中力作用的悬臂圆环曲梁平面弯曲的三维分析	173
4.8 均匀拉力作用下含圆孔板的孔边应力集中	188
4.9 两端固定的厚壁管道在自重作用下的变形和应力	204

第5章 膜和薄壳问题	213
5.1 含椭圆孔的椭圆薄膜在外部张力作用下的静力分析	213
5.2 圆形薄膜大变形静力分析	223
5.3 柱形容器在内压作用下的静力分析	229
5.4 圆柱形薄壳在均匀内压作用下的静力分析	233
第6章 板的弯曲和壳体计算	238
6.1 简支和固支圆板的在不同荷载作用下的弯曲	238
6.2 悬臂长板的大挠度弯曲	249
6.3 用壳体单元分析受均布荷载作用的固支圆板大挠度弯曲	261
6.4 利用拉伸操作建立膨胀弯管模型	266
6.5 两端简支开口柱壳在自重作用下的静力分析	275
6.6 圆筒在一对横向集中力作用下的变形	281
6.7 两边简支开口柱壳在集中力作用下的大变形曲屈	290
第7章 简单振动系统	299
7.1 单自由度弹簧质量系统的频率计算	299
7.2 悬索自由振动的频率	303
7.3 用弹簧单元连接的圆盘的扭转振动	308
7.4 圆杆连接圆盘的扭转振动	312
7.5 钻杆的扭转自由振动	316
第8章 梁的振动分析	323
8.1 简支梁的自振频率计算	323
8.2 自由-自由梁的纵向自由振动	333
8.3 有轴向压力作用的简支梁的自由振动	337
8.4 用壳体单元计算悬臂等强度梁的自由振动	343
8.5 矩形截面薄壁悬臂梁的自由振动	349
第9章 膜板和实体振动	356
9.1 圆形张紧薄膜的自由振动	356
9.2 薄膜二维非轴对称自由振动分析	362
9.3 薄膜三维非轴对称振动分析	368
9.4 悬臂长板的自由振动频率	376
9.5 悬臂宽板的模态分析	380
9.6 固支圆板的自由振动	386
9.7 用实体单元分析圆环的振动	390
9.8 机翼模型的振动分析	395

第10章 平面建模分析和三维实体建模	403
10.1 带三个圆孔的平面支座分析	403
10.2 角支座应力分析	408
10.3 立体斜支座的实体建模	414
10.4 四分之一车轮实体建模	420
10.5 轴承支座的实体建模	423
第11章 最优化设计	432
11.1 最优化设计绪论	432
11.2 最优化问题框架	433
11.3 ANSYS优化设计流程	434
11.4 变截面悬臂梁的外形形状优化	434
11.5 平面刚架的优化设计	445
第12章 层合板和断裂力学	452
12.1 四边简支方形层合板在均布外载作用下的变形	452
12.2 均布拉力作用下含裂纹板的应力强度因子计算	464

第2篇 工程应用篇

第13章 用APDL实现 空间网壳结构参数化建模	483
13.1 K系列球面网壳结构特点和建模	483
13.1.1 K系列球面网壳的特点	483
13.1.2 几何描述	484
13.1.3 杆件连接关系	484
13.2 参数化设计语言APDL介绍	484
13.2.1 参数和表达式	485
13.2.2 ANSYS中的基本指令	485
13.2.3 分支和循环	486
13.3 用户界面设计语言UIDL介绍	488
13.3.1 单行参数输入	488
13.3.2 多行参数输入	488
13.4 网壳建模程序设计	489
13.4.1 模型建立的步骤	489
13.4.2 节点坐标计算	489
13.4.3 单元连接	491
13.4.4 变量说明	492
13.4.5 节点坐标计算公式	492
13.4.6 主框图说明	493

13.4.7 单元连接关系定义	493
13.4.8 源程序	494
13.5 程序使用说明	497
13.5.1 加载程序	497
13.5.2 界面说明	497
13.5.3 注意事项	499
13.6 应用举例	499
13.6.1 基本参数	499
13.6.2 输入数据并生成模型	499
13.6.3 输入单元参数和荷载后开始计算	500
13.6.4 选择结果输出方式	500
第14章 塔式起重机静动力分析	501
14.1 塔式起重机基本概念	501
14.2 塔式起重机拓扑模型	501
14.3 塔机模型受力分析	503
14.3.1 部件受力特征分析	504
14.3.2 截面参数定义	504
14.3.3 自重荷载和配重	504
14.3.4 选用合适的分析模型。	504
14.3.5 固定塔身底部的4个节点	505
14.4 塔机建模程序设计	505
14.4.1 塔身节点计算和单元连接	505
14.4.2 塔顶建模	506
14.4.3 塔臂建模	506
14.4.4 平衡臂和斜拉索建模	507
14.5 塔机静力分析	508
14.6 塔机模态分析	514
14.7 塔机静动力分析程序	517
第15章 长柱形天然气罐在内压作用下的静力分析	534
15.1 概况	534
15.2 建立模型	534
15.3 利用轴对称壳单元SHELL51计算	535
15.3.1 单元基本性质和约定	535
15.3.2 求解过程	537
15.3.3 源程序	540
15.3.4 计算结果	542

目 录

15.3.5	筒体部分理论解.....	544
15.3.6	结果讨论.....	544
15.4	利用8节点2D实体单元PLANE82单元计算.....	548
15.4.1	建立模型.....	549
15.4.2	计算过程.....	549
15.4.3	计算结果及讨论.....	556
15.4.4	源程序.....	557
15.5	用20节点3D实体单元solid95计算1/4模型.....	558
15.5.1	建立1/4三维模型.....	558
15.5.2	计算步骤.....	559
15.5.3	计算结果分析.....	565
15.5.4	与弹性力学解答的对比.....	568
15.5.5	计算程序.....	569
附录1	常用结构单元参考.....	571
附录2	结构分析命令速查.....	604

Ansys 应用实例样章

第1篇 教学实例篇

第1章 简单拉压杆结构

杆系结构是指结构由许多细长杆件构成的结构系统，杆件的弯曲刚度较小，或者弯曲产生的应力和轴力相比较小，每个杆件的主要变形是轴向拉伸和压缩变形。对于这一类问题，有限元模型可以利用杆单元模型（LINK）来处理。在ANSYS中，二维杆单元是LINK1，三维杆单元是LINK8。它们的单元两端的节点位移分别有2个和3个线位移。对于许多杆系空间结构需要利用LINK8单元求解，计算结果除了节点位移外，我们最关心的是各个杆件的内力和应力。

在ANSYS中杆件的内力无法直接得到，需要利用定义单元表（ETABLE）的方法获得。求解每个有限元问题的基本过程是一样的：首先建立模型并施加力和位移边界条件，然后就可以求解了。在求解之后得到节点位移，其余的力学量需要通过适当的定义或者运算才可以得到。

杆系结构的计算主要得到桁架结构的变形和内力、反力。对于复杂的杆系结构，目前还没有简便的建模方法，通常还是需要利用节点和单元的定义指令N（Node）和E（Element）来定义。这种建立模型的方式叫做单元直接建模，它的主要优点是直观。对于边界条件：荷载由命令F（Force）来定义，位移约束由命令D（Displacement）来定义。

当计算模型建立好，位移边界条件定义妥当，荷载施加完毕后，就可以发出求解命令SOLVE通知系统开始求解。系统对每一个单元计算刚度矩阵后，叠加生成总体刚度矩阵，生成节点荷载向量，通过引入位移边界条件修正总体刚度矩阵和荷载向量后，开始求解位移方程而得到各个节点的位移值。再次调用单元刚度矩阵计算各个单元的内力、各个位移约束处的反力等。

得到这些计算结果后，进入后处理模块Post1可以显示结果和观察变形、应力分布等情况。变形图通常用PLDISP（PLot DISPlacement）来显示。节点上的计算结果，如内力和应力可以用PLNSOL（Plot Node SOLution）来完成。不同类型的单元具有不同的内力和应力约定，单元内的计算结果通过PLESOL（Plot Element SOLution）命令来实现。

以上是结构静力分析的大体过程。一般的动态分析主要是模态分析，就是通过对刚度矩阵和质量矩阵的迭代计算，得到结构的前几阶振动频率和振型。下面列出的是进行分析要用到的最基本的命令：

1. 最基本命令

（1）建立模型（/Prep7模块命令）。

通过节点的坐标位置定义节点：N, NODE, X, Y, Z, THXY, THYZ, THZX；在 (X,Y,Z) 坐标位置定义第NODE号节点，节点坐标系方向为THXY, THYZ和THZX。

通过节点连接关系定义单元：E, I, J, K, L, M, N, O, P：过节点I~P定义单元。

(2) 定义边界条件(/Solu模块命令)

定义节点上的位移约束：D, NODE, Lab, VALUE, VALUE2, NEND, NINC, Lab2, Lab3, Lab4, Lab5, Lab6：对节点NODE施加位移约束。

定义作用在节点上的外力：F, NODE, Lab, VALUE, VALUE2, NEND, NINC：给节点NODE施加节点力。

(3) 求解方程(/Solu模块命令)。

开始求解计算：SOLVE。

(4) 查看结果(/post1模块命令)。

显示结构变形图：PLDISP,2。

显示节点结果：PLNSOL, Item, Comp。

2. 杆单元特性

杆单元只承受轴向力，单元的内力主要是轴力。除此之外，还有轴向应力和轴向应变，杆单元的两端的节点只有线位移自由度。对于二维杆单元LINK1，每个单元的两端有两个节点，每个节点有两个线位移。对于三维杆单元LINK8，每个单元也有两个节点，但每个节点有3个线位移自由度。

单元输入的几何参数只有杆件的截面面积A，材料参数有弹性模量EX，热膨胀系数ALPX，密度DENS和阻尼DAMP。杆件单元的定义通过用E指令定义两个端点的节点编号即可，当前单元的单元类型使用当前默认值，该默认值可以通过指令TYPE来改变。当前单元的材料参数使用当前的默认值，该默认值可以通过指令MAT来改变。当前单元的实常数使用当前的默认值，该默认值可以通过指令REAL来改变。

1.1 铰接杆在外力作用下的变形

在两个相距 $a=10\text{m}$ 的刚性面之间，有两根等截面杆铰结在2号点，杆件与水平面的夹角 $=30^\circ$ ，在铰链处有一向下的集中力 $F=1000\text{N}$ ，杆件材料的弹性模量 $E=210\text{GPa}$ ， $A=10\text{cm}^2$ （如图1.1所示），试利用二维杆单元LINK1确定两根杆件内力和集中力位置处的位移。杆件变形很小，可以按照小变形理论计算。

1. 理论解

这是一个静定结构，两根杆的内力可以很容易地用理论力学知识求解出来。考虑杆件变形时，杆件伸长量和节点位移之间的关系也不复杂。由于结构几何形状和受力是左右对称的，结构的变形特征也是对称的，杆件伸长量与节点位移的关系就更加简单了。

根据2号节点处的平衡关系，很容易得到两个拉杆的拉力都等于 1000N ，由于杆件的截面积等于 1000mm^2 ，所以杆件内的拉应力等于 1MPa 。




图1.1 两杆桁架结构模型图

杆件的实际长度为 $l = 5 / \cos 30^\circ$ ，根据虎克定律：

$$\Delta l = \frac{Nl}{EA} = \frac{1000 \times 5000 / \cos 30^\circ}{210 \times 10^3 \times 1000} = 0.02749 \text{ mm}$$

根据2号节点处的位移和杆件伸长量的关系，竖向位移应该是伸长量的2倍，所以，竖向位移的理论值应该是0.05498mm。

2. ANSYS求解

该结构包含两根杆件，所以可以划分为两个单元。在左侧的刚性墙壁上设置节点1，中间两根杆件连接点设节点2，右侧刚性墙壁上的固定点设为节点3。它们的编号和坐标位置如图1.1所示，即1 (0,0)，2 (a/2, -b)，3 (a,0)。1号节点和3号节点固定，2号节点上有集中力作用。

使用节点定义命令N (Node) 可以完成对这些节点的定义。用命令ET (Element Type) 申明该系统使用二维杆单元LINK1，用实常数定义命令R (Real constant) 定义杆件的横截面面积，用材料参数定义命令MP (Material Property) 定义材料的弹性模量。用位移约束命令D (Displacement) 固定节点，用F (Force) 施加节点力。用Solve命令开始求解，得到节点位移结果后，可以用PLDISP (Plot DISplacement) 绘制结构变形图，或者PRDISP列表显示节点位移结果。用定义单元表命令ETABLE (Element TABLE)，可以提取杆件轴向应力。这些命令构成了下面的主要求解步骤。

(1) 设置工程选项，定义几何参数，定义变量。

(2) 定义节点。

进入前处理模块，用定义节点的命令N定义，在1 (0,0)，2 (a/2, -b)，3 (a,0) 位置的3个节点。

(3) 定义单元。

用单元类型定义命令ET定义第1类单元为二维杆单元LINK1。用单元定义命令E (Element) 定义单元1与节点1和节点2连接，单元2与节点2和节点3连接。

(4) 位移约束和荷载定义。

用位移约束命令D固定1号节点和3号节点，用F (Force) 在2号节点施加沿着-Y轴方向的集中力。在ANSYS环境下定义的结构模型如图1.2所示。

(5) 求解节点位移。

进入求解模块，用指令Solve开始求解，获得节点位移结果。

(6) 用单元表提取杆件轴力。

用PLDISP绘制变形后的结构示意图（如图1.3所示）。用“*GET”命令提取2号节点的竖向位移，用ETABLE定义轴力单元表，并提取左边杆件的轴力。

(7) 将计算结果保存到结果文件。

下面给出了这些步骤的源程序，感叹号后面是当前行的注释。




图1.2 建立的两杆桁架结构模型图




图1.3 两杆桁架在集中力作用下的变形图

3. ANSYS程序

```

FINISH
/CLEAR,NOSTART      ! 清除已有的数据
! (1) 设置工程的选项，定义几何参数
/FileName, EX1.1      ! 定义工程文件名称
/TITLE, EX1.1, DEFLECTION OF A HINGED SUPPORT
*AFUN,DEG            ! 将三角函数的默认单位由弧度改为角度
THETA=30              ! 定义变量Theta为杆件与水平面的夹角30度
A=10000               ! 定义变量A为节点3的水平x坐标10000mm
B=A/2*TAN(THETA)      ! 节点2的铅直y坐标
! (2) 进入前处理模块，定义节点
/PREP7                ! 加载前处理模块
N,1                   ! 在坐标原点(0,0,0)定义第1号节点
N,2,A/2,-B            ! 在 (A/2,-B) 定义第2号节点
N,3,A                 ! 在 (A,0) 定义第3号节点
! (3) 在前处理模块中,定义单元类型、材料参数和各个单元
ET,1,LINK1            ! 定义第1类单元为二维杆单元LINK1
R,1,1000              ! 定义第1类实常数, 杆件截面面积A=1000mm^2
MP,EX,1,210E3         ! 定义第1类材料的弹性模量EX = 210000N/mm^2
E,1,2                 ! 通过节点1,2定义单元1
E,2,3                 ! 通过节点2,3定义单元2

```



```

! (4) 在前处理模块中,定义位移约束
D,1,ALL,,,3,2      ! 约束1号节点的所有位移自由度,按增量2循环到3号节点来约束3号节点
F,2,FY,-1000       ! 在2号节点上施加沿着-Y方向大小为1000N的集中力
OUTPR,,1           ! 输出第1个荷载步的基本计算结果
FINISH             ! 退出前处理模块PREP7

! (5) 进入求解模块,开始求解
/SOLU              ! 加载求解模块
SOLVE              ! 开始静力分析的求解
FINISH             ! 退出求解模块SOLU

! (6) 进入后处理模块,显示节点位移和杆件内力
/POST1            ! 加载后处理模块
PLDISP,2          ! 显示结构变形图,参数“2”表示用虚线绘制出原来结构的轮廓
MID_NODE = NODE (A/2,-B,0) ! 寻找距离位置(A/2,-B,0)最近的节点,存入MID_NODE
*GET,DISP,NODE,MID_NODE,U,Y ! 提取节点MID_NODE上的位移UY
LEFT_EL = ENEARN (MID_NODE) ! 寻找距离节点MID_NODE最近的单元,存入LEFT_EL
ETABLE,STRS,LS,1    ! 用轴向应力SAXL的编号“LS,1”定义单元表STRS
*GET,STRSS,ELEM,LEFT_EL,ETAB,STRS ! 从单元表STRS中提取LEFT_EL单元的应力结果,存入变量STRSS

! (7) 申明数组,提取计算结果,并比较计算误差
*DIM,LABEL,CHAR,2   ! 定义2个元素的字符型数组LABEL
*DIM,VALUE,,2,3     ! 定义2*3的数值型数组VALUE
LABEL(1) = 'STRS_MPa','DEF_mm' ! 给字符型数组的第1个元素赋值
*VFILL,VALUE(1,1),DATA,1,-0.05498 ! 给其他数值型数组中的元素赋理论值
*VFILL,VALUE(1,2),DATA,STRSS,DISP
*VFILL,VALUE(1,3),DATA,ABS(STRSS /1 ),ABS( DISP /0.05498 )
/OUT,EX1_1,out      ! 将输出内容重定向到文件EX1_1.out
/COM               ! 以注释形式输出内容
/COM,----- EX1.1 RESULTS COMPARISON -----
/COM,
/COM,      |  TARGET  |  ANSYS  |  RATIO
/COM,
*VWRITE,LABEL(1),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,' ',F10.3,' ',F10.3,' ',1F5.3)
/COM,-----
/OUT          ! 结束数据重定向,关闭输出文件
FINISH        ! 退出后处理模块
*LIST,EX1_1,out ! 列表显示文件EX1_1.out的内容

```

4. 结果分析

经过计算,ANSYS在计算结果文件EX1_1.txt中给出了节点位移计算结果-0.055mm,杆件轴力1MPa,和前面的理论解完全一致。

1.2 人字形屋架的静力分析

跨度8m的人字形屋架，左边端点是固定铰链支座，右端是滑动铰链支座。在上面的3个节点上作用有3个向下集中力 $P=1\text{kN}$ ，结构的几何尺寸和边界条件如图1.4所示，试利用二维杆单元LINK1分析该屋架在3个集中力作用下的变形和内力。

1. 理论解

在小变形假设下，两个支座间不会出现水平反力。整个结构和荷载左右对称，所以结构的内力也是左右对称的。只分析左面一半即可知道整个结构的受力了。整个结构是一个静定结构，所以各个杆件的内力可以利用理论力学中的节点法，或者截面法求解出来。依据整个结构的平衡，可以得到左右两个支座的竖向反力为 1.5kN 。

按照图1.5中的节点编号，根据1号和5号单元在1号节点处的平衡，可以确定出1号单元的内力是 $N_1=3\text{kN}$ ，5号单元的内力是 $N_5=1.5\sqrt{5}\text{kN}$ 。从3号节点的平衡可以知道，9号单元的内力为0。

取左面一半结构，对节点7列出力矩平衡方程：

$$M_7(F)=0, -1.5\times 4+1\times 2+N_2\times 2=0$$

可以求出2号单元的内力为 $N_2=2\text{kN}$ 。再根据2号节点处的平衡方程：

$$X=0, N_2-N_1+N_8/\sqrt{2}=0$$

$$Y=0, N_7+N_8/\sqrt{2}=0$$

可以解出： $N_8=\sqrt{2}\text{kN}$ ， $N_7=-1\text{kN}$

再由6号节点的平衡方程：

$$X=0, N_6\times 2/\sqrt{5}-N_5\times 2/\sqrt{5}=0$$

可以解出： $N_6=1.5\sqrt{5}\text{kN}$

至此，我们得到所有杆件的内力为：

$$N_1=N_4=3\text{kN}, N_2=N_3=2\text{kN}, N_5=N_{13}=1.5\sqrt{5}\text{kN}, N_6=N_{12}=1.5\sqrt{5}\text{kN},$$

$$N_7=N_{11}=-1\text{kN}, N_8=N_{10}=\sqrt{2}\text{kN}, N_9=0。$$

如果需要计算结构在荷载作用下的变形，就必须使用材料力学的知识，且求解过程比较复杂。

2. ANSYS求解

首先将这个结构在杆件相交的地方，设置节点，节点之间用单元相连接（如图1.5所示）。利用节点定义命令N定义7个节点，用单元定义命令E定义13个单元；用荷载定义命令F在模型的6，7，8节点上施加集中荷载，用位移约束命令D定义1号节点两个位移全部被约束，5号节点的竖向位移被约束。利用下面提供的程序可以完成对该结构的变形和内力计算。

如图1.6所示给出的是使用ANSYS程序的步骤（1）到（3）生成好的有限元模型；在第（4）步用D命令定义位移约束，F命令定义荷载后，用SOLVE求解得到各个节点的位移。第（5）步进入一般后处理模块，用PLDISP图形显示结构变形图（如图1.7所示）。在第（6）

步用ETABLE定义单元表，以提取单元的计算结果：轴力，轴向应力和轴向应变。在用PRETAB命令列表显示单元表内容的同时，图1.8所示是用PLETAB命令以色度图方式表示的杆件轴力分布情况。




图1.4 人字形屋架计算模型示意图




图1.5 人字形屋架的单元划分图




图1.6 人字形屋架的结构模型图




图1.7 屋架在屋面荷载作用下的变形




图1.8 屋架杆件的轴力等高线色度图

3. ANSYS程序

```
FINISH ! 退出以前的模块
/CLEAR, START ! 清除系统中的所有数据, 不读入启动文件的设置
! (1) 设置工程的选项
/FILNAME, EX1.2 ! 指定当前工程的文件名, 所有数据用该文件名保存
```

```

/UNITS, SI      ! 申明采用国际单位制
/TITLE, EX1.2, Plane Roof Tuss Model. ! 定义问题的标题
! (2) 进入前处理模块, 定义节点
/PREP7      ! 进入前处理模块: 定义模型
N, 1, 0, 0    ! 通过节点的坐标位置定义各个节点, 1号节点1(0, 0)
N, 2, 2, 0    ! 2号节点2(2m, 0)
N, 3, 4, 0    ! 3号节点3(4m, 0)
N, 4, 6, 0    ! 4号节点4(6m, 0)
N, 5, 8, 0    ! 5号节点5(8m, 0)
N, 6, 2, 1    ! 6号节点6(2m, 1m)
N, 7, 4, 2    ! 7号节点7(4m, 2m)
N, 8, 6, 1    ! 8号节点8(6m, 1m)
! (3) 在前处理模块中, 定义单元类型、材料参数和各个单元
ET, 1, LINK1   ! 定义第1类单元为二维杆单元LINK1
MP, EX, 1, 207E9 ! 定义第1类材料弹性模量EX = 207GPa
R, 1, 0.01     ! 定义杆件第1类实常数, 截面积0.01m^2
E, 1, 2        ! 通过节点连接关系定义各个单元, 单元1由1, 2节点组成
E, 2, 3        ! 单元2由2, 3节点组成
E, 3, 4        ! 单元3由3, 4节点组成
E, 4, 5        ! 单元4由4, 5节点组成
E, 1, 6        ! 单元5由1, 6节点组成
E, 6, 7        ! 单元6由6, 7节点组成
E, 2, 6        ! 单元7由2, 6节点组成
E, 2, 7        ! 单元8由2, 7节点组成
E, 3, 7        ! 单元9由3, 7节点组成
E, 4, 7        ! 单元10由4, 7节点组成
E, 4, 8        ! 单元11由4, 8节点组成
E, 7, 8        ! 单元12由7, 8节点组成
E, 8, 5        ! 单元13由8, 5节点组成
/PNUM, NODE, 1 ! 打开节点编号显示
/NUMBER, 2     ! 只显示编号, 不使用彩色
/PNUM, ELEM, 1 ! 打开单元编号显示开关
EPLOT          ! 图形显示单元
FINISH        ! 退出前处理模块
! (4) 进入求解模块, 设置求解选项, 设置位移边界条件, 加载并开始求解
/SOLU          ! 进入求解模块: 定义力和位移边界条件, 并求解
ANTYPE, STATIC ! 申明分析类型是静力分析 (STATIC或者0)
OUTPR, BASIC, ALL ! 在输出结果中, 列出所有荷载步的基本计算结果
D, 1, ALL, 0    ! 约束1号节点的所有节点位移自由度UX和UY
D, 5, UY, 0     ! 约束5号节点的竖向自由度UY
F, 6, FY, -1000 ! 对6号节点施加-Y方向的集中力1000N
F, 7, FY, -1000 ! 对7号节点施加-Y方向的集中力1000N
F, 8, FY, -1000 ! 对8号节点施加-Y方向的集中力1000N
SOLVE          ! 发出求解指令

```

```

FINISH          ! 退出求解模块
! (5) 进入一般后处理模块, 显示结构变形图
/POST1          ! 进入一般后处理模块: 显示变形和内力计算结果
PLDISP, 2       ! 显示结构变形图, 保留未变形结构的轮廓
! (6) 列表显示节点位移和单元的计算结果
PRDISP          ! 列表显示节点位移值计算结果
ETABLE, MFORX, SMISC, 1      ! 以杆单元的轴力为内容, 建立单元表MFORX
ETABLE, SAXL, LS, 1          ! 以杆单元的轴向应力为内容, 建立单元表SAXL
ETABLE, EPELAXL, LEPEL, 1    ! 以杆单元的轴向应变为内容, 建立单元表EPELAXL
PRETAB          ! 显示单元表中的计算结果
/NUMBER, 0      ! 显示编号, 并使用彩色
PLETAB, MFORX   ! 用色度图显示单元表MFORX中杆件轴力图
FINISH         ! 退出后处理模块

```

4. 结果分析

通过建立模型的步骤(1)至(3), 得到几何模型, 要在模型上显示节点编号, 需要用编号显示命令“/PNUM, NODE, 1”打开节点编号, 关闭显示需要将这里的“1”该为“0”。命令“/PNUM, ELEM, 1”可以打开单元编号显示, 默认的单元颜色显示是编号和彩色同时使用, 即各个单元具有不同的颜色。为了突出显示效果, 这里使用了选项“/NUMBER, 2”将该设置修改为只显示编号, 不使用颜色。后面第(6)步由命令“/NUMBER, 0”恢复为彩色和编号同时使用, 以突出彩色等高线图的显示。

施加荷载和边界条件后, 模型图中会出现荷载符号和节点约束标志。这里施加在节点上的集中力一般用红色箭头表示, 每一个位移约束用一个小三角符号表示。所以在左侧节点由两个小三角, 表示两个方向的线位移被约束。右侧支座则只有一个小三角, 表示只约束了竖向线位移。

通过求解后, 第(6)步利用PRDISP列出了各个节点的位移计算结果(表1.1)。

表1.1 人字型屋架的节点位移计算结果

节 点 号	水平位移UX(m)	铅直位移UY(m)
1	0.0000	0.0000
2	0.28986E-05	-0.26666E-04
3	0.48309E-05	-0.25865E-04
4	0.67633E-05	-0.26666E-04
5	0.96618E-05	0.0000
6	0.95235E-05	-0.27149E-04
7	0.48309E-05	-0.25865E-04
8	0.13832E-06	-0.27149E-04

用定义单元表的方法, 可以提取出各个单元的内力和应力。用列出单元表的计算结果指令PRETAB可以查看轴力、轴向应力和轴向应变的计算结果; 表1.2是由PRETAB列出的各个单元的轴力、轴向应力和轴向应变计算结果; 表1.3是这几项计算结果的最小值和所在

单元；表1.4是这几项计算结果的最大值和所在单元。

表1.2 所有单元内力计算结果

单元编号	轴力MFORX	轴向应力SAXL	轴向应变EPELAXL
1	3000.0	0.30000E+06	0.14493E-05
2	2000.0	0.20000E+06	0.96618E-06
3	2000.0	0.20000E+06	0.96618E-06
4	3000.0	0.30000E+06	0.14493E-05
5	-3354.1	-0.33541E+06	-0.16203E-05
6	-3354.1	-0.33541E+06	-0.16203E-05
7	-1000.0	-0.10000E+06	-0.48309E-06
8	1414.2	0.14142E+06	0.68319E-06
9	0.0000	0.0000	0.0000
10	1414.2	0.14142E+06	0.68319E-06
11	-1000.0	-0.10000E+06	-0.48309E-06
12	-3354.1	-0.33541E+06	-0.16203E-05
13	-3354.1	-0.33541E+06	-0.16203E-05

表1.3 轴力、向应力和轴向应变的最小值所在单元

所在单元ELEM	13	13	13
最小值	-3354.1	-0.33541E+06	-0.16203E-05

表1.4 轴力、向应力和轴向应变的最大值及所在单元

所在单元ELEM	1	1	1
最大值	3000.0	0.30000E+06	0.14493E-05

和前面的理论计算结果比较，内力计算结果吻合很好。

1.3 超静定拉压杆的反力计算

在两个相距 $l=1\text{m}$ 刚性面之间有一根等截面杆，杆件材料的弹性模量为 $E=210\text{GPa}$ ，在距离上端 $a=0.3\text{m}$ 和 0.6m 截面位置处分别为受到沿杆件轴向的两个集中力 $F_1=1000\text{N}$ 和 $F_2=500\text{N}$ （如图1.9所示），试利用二维杆单元LINK1确定两个刚性面对杆件的支反力 R_1 和 R_2 。

1. 理论解

这是一个一次超静定结构，杆件在变形前和变形后的长度不变。按图1.9所示的反力作用情况，1号单元的轴力为 $-R_2$ ，2号单元的轴力为 F_2-F_1 ，3号单元的轴力为 R_1 ，则这3个单元的伸长量分别为：

$$\Delta l_1 = \frac{0.4R_2}{EA}, \quad \Delta l_2 = \frac{0.3F_2 - 0.3R_2}{EA}, \quad \Delta l_3 = \frac{0.3R_1}{EA}$$

根据总长度不变可以得到：

$$\Delta l_1 + \Delta l_2 + \Delta l_3 = \frac{-0.7R_2 + 0.3F_2 + 0.3R_1}{EA} = 0$$

平衡关系：

$$R_1 - F_1 - F_2 + R_2 = 0$$

由此解出，反力的结果为：

$$R_1 = 900\text{N}, R_2 = 600\text{N}$$

2. ANSYS求解

在两个集中力位置和两个刚性固定位置设置4个节点，杆件划分为3个单元。坐标系，节点和单元的编号如图1.10所示。按下面步骤完成分析。




图1.9 超静定拉压杆的模型示意图



图1.10 节点和单元划分

- (1) 首先申明分析类型是静力分析，单元类型采用二维杆单元LINK1，定义横截面A和材料的弹性模量EX。
- (2) 利用N命令定义这4个节点。
- (3) 用E命令定义中间的3个单元。
- (4) 定义上端1号点和下端4号点固定，并施加在2号点和3号点的集中力。
- (5) 指定输出选项后，开始求解。
- (6) 在一般后处理模块中，利用节点力求和操作获得两个固定点处的反力。
- (7) 将计算结果输出到计算结果文件。

3. ANSYS程序

```
FINISH          ! 退出以前的模块
/CLEAR,START    ! 清除系统中的所有数据,读入启动文件的设置
! (1) 设置工程的选项, 进入前处理模块,声明单元类型,实常数和材料参数
/FILNAME, EX1.3 ! 指定当前工程的文件名,所有数据用该文件名保存
/PREP7          ! 加载前处理模块
/TITLE, EX1.3, STATICALLY INDETERMINATE REACTION FORCE ANALYSIS
```

```

ANTYPE,STATIC      ! 定义分析类型为静力分析
ET,1,LINK1         ! 定义第1类单元为二维杆单元LINK1
R,1,1              ! 定义实常数,杆件横截面面积为1
MP,EX,1,210E9      ! 定义第1类材料的弹性模量为210GPa
! (2) 定义节点
N,1                ! 在坐标原点定义第1号节点
N,2,,0.4           ! 在(0,0.4m)位置定义2号节点
N,3,,0.7           ! 在(0,0.7m)位置定义3号节点
N,4,,1.0           ! 在(0,1.0m)位置定义4号节点
! (3) 定义单元
E,1,2              ! 过节点1,2定义单元
EGEN,3,1,1         ! 按前面单元的节点连接情况,从单元1开始生成3个单元,每次节点码增加1
! (4) 定义位移约束和荷载
D,1,ALL,,,4,3      ! 定义位移边界条件,固定1号节点,使用增量3循环到4,固定4号节点
F,2,FY,-500        ! 在2号节点处施加沿Y负方向大小为500N的集中力
F,3,FY,-1000       ! 在3号节点处施加沿Y负方向大小为1000N的集中力
FINISH             ! 退出前处理模块
! (5) 进入求解模块SOLU
/SOLU              ! 加载求解模块
OUTPR,BASIC,1      ! 选择基本输出项目
OUTPR,NLOAD,1      ! 指定输出第1荷载步的内容
SOLVE              ! 开始求解
FINISH             ! 退出求解模块
! (6) 进入一般后处理模块POST1
/POST1             ! 进入一般后处理模块
NSEL,S,LOC,Y,1.0    ! 选择所有位于Y=1.0位置上的节点,即节点4
FSUM               ! 累计叠加选择集中(即4号节点)所有节点上的反力
*GET,REAC_1,FSUM,,ITEM,FY ! 将累加结果中的FY(Y方向的力)保存到变量REAC_1中
NSEL,S,LOC,Y,0      ! 选择所有位于Y=0位置上的节点,即节点1
FSUM               ! 累加选择集中(即1号节点)所有节点上的力
*GET,REAC_2,FSUM,,ITEM,FY ! 将累加结果中的FY(Y方向的力)保存到变量REAC_2中
! (7) 申明数组,输出计算结果并比较
*DIM,LABEL,CHAR,2    ! 定义两个字符型数组LABEL
*DIM,VALUE,,2,3      ! 定义2*3数值型数组VALUE
LABEL(1) = 'R1, N ', 'R2, N ' ! 给数组LABEL的第1列元素赋值
*VFILL,VALUE(1,1),DATA,900,600 ! 给数组VALUE的第1列元素赋理论值
*VFILL,VALUE(1,2),DATA,ABS(REAC_1),ABS(REAC_2) ! 给数组VALUE的第2列元素赋计算值
*VFILL,VALUE(1,3),DATA,ABS(REAC_1/900),ABS(REAC_2/600) ! 赋比值
/OUT,EX1_3,out       ! 将输出内容重定向到文件EX1_3.out
/COM
/COM,----- EX1.3 RESULTS COMPARISON -----
/COM,
/COM,      | TARGET | ANSYS | RATIO
/COM,

```



```

*VWRITE,LABEL(1),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,' ',F10.1,' ',F10.1,' ',1F5.3)
/COM,-----
/OUT          ! 结束数据重定向,关闭输出文件
FINISH        ! 退出后处理模块
*LIST,EX1_3,out ! 列表显示文件EX1_3.out的内容

```

4. 结果分析

利用上面程序,计算得到表1.5的有限元结果和理论结果的比较,从中可以看出,两者完全吻合。

表1.5 上下固定端处反力的理论计算结果和有限元计算结果的比较

反 力	理 论 解	ANSYS解	比 值
上端反力R1 (N)	900.0	900.0	1.000
下端反力R2 (N)	600.0	600.0	1.000

1.4 平行杆件与刚性梁连接的热应力问题

长度 $l=1\text{m}$ 的两根铜杆和一根铁杆,平行地铰结在刚性壁和刚性梁之间(如图1.11所示),它们的横截面积都等于 $A=65\text{mm}^2$,铜和铁的弹性模量分别等于 $E_c=100\text{GPa}$ 和 $E_s=210\text{GPa}$,热膨胀系数分别等于 $\alpha_c=1.6\times 10^{-5}/^\circ\text{C}$ 和 $\alpha_s=1.3\times 10^{-5}/^\circ\text{C}$ 。刚性壁上铰节点之间的距离为 1m ,试计算在刚性梁中部的集中力 $Q=4000\text{N}$ 作用下,温度升高 $T=10^\circ\text{C}$ 以后,3根杆件中的应力。




图1.11 外力和温度作用下的平行杆刚梁模型




图1.12 有限元节点和单元布置

1. 理论解

刚性梁连接保证了这3根杆在变形前长度相等,升温受力后它们的长度仍然成比例。但是,由于整个结构几何和受力左右对称,所以变形也是左右对称的。

假设刚性梁在变形后向下移动了 δ ,则对于铜杆和铁杆分别有:

$$\delta = \frac{N_c l}{E_c A} + \alpha_c \Delta T l \quad \delta = \frac{N_s l}{E_s A} + \alpha_s \Delta T l$$

其中, N_c , N_s 分别是铜杆和铁杆中的轴力。根据刚性梁上力的平衡关系,有:

$$2N_c + N_s - Q = 0$$

由这三个方程可以解出刚性梁的下降量，铜杆和铁杆的拉力为：

$$N_c = \frac{E_c}{2E_c + E_s} [(\alpha_s - \alpha_c)E_s A \Delta T + Q] \quad N_s = \frac{E_s}{2E_c + E_s} [2(\alpha_c - \alpha_s)E_c A \Delta T + Q]$$

代入数值，得到轴力：

$$N_c = 875.73 \text{ N}, N_s = 34.59 \text{ N}$$

所以，铜杆和铁杆中的轴向应力分别为：

$$\sigma_c = 875.73/65 = 13.74 \text{ MPa}, \sigma_s = 2248.5/65 = 34.59 \text{ MPa}$$

2. ANSYS求解

根据如图1.12所示的有限元模型，节点1，2，3固定，节点4，5，6是连接在刚性梁上的铰链节点上。




图1.13 平行杆刚性梁问题的有限元模型

(1) 定义分析类型、单元类型、实常数和材料参数。

在前处理模块中，定义分析类型ANTYPE是静力分析（STATIC），单元类型ET是二维杆单元LINK1，实常数截面面积R为65mm²，材料参数（MP）中的弹性模量EX=210GPa，热膨胀系数ALPX等。

(2) 定义节点位置信息。

利用节点定义命令N，先在1（-1m, 0）和3（1m, 0）处定义1号和3号节点，利用FILL命令填充生成介于1号和3号之间（0, 0）位置的2号节点；同样地可以定义4（-1m, -1m），6（1m, -1m）节点，和节点5（0, -1m）。

(3) 定义单元信息。

用“E,1,4”定义过节点1，4节点的铜杆单元（使用默认的单元类型和材料类型）；用“E,3,6”定义过节点3，6节点的铜杆单元。用“MAT,2”将材料类型号改变为2，用“E,2,5”定义过节点2，5节点的铁杆单元。用“CP,1,UY,5,4,6”定义5，4，6这3个节点为耦合自由度，即3者的UY位移总是相等。

(4) 定义位移约束和荷载（外力和温度荷载）。

用“D,1,ALL,,,3”定义1号节点到3号节点全部固定；用“F,5,FY,-4000”在5号节点处

施加沿着-Y方向的大小等于4000N的集中力；用“BFUNIF,TEMP,80”将温度从原来的70度均匀升到80度(TREF+10)。经过以上步骤得到如图1.13所示的有限元模型。

(5) 进入求解模块，定义求解选项，并开始求解。

用“/SOLU”加载求解模块；用“SOLVE”命令开始求解，获得各个节点的位移计算结果。

(6) 在一般后处理模块中，寻找特定位置的节点和单元，并提取它们的内力。

用“/POST1”加载后处理模块；用“ETABLE,STRS_ST,LS,1”以杆件的轴向应力“LS,1”为内容定义单元表STRS_ST；用“ETABLE,STRS_CO,LS,1”以杆件的轴向应力“LS,1”定义单元表STRS_CO。用“*GET,STRSS_ST,ELEM,STEEL_E,ETAB,STRS_ST”从单元表STRS_ST中提取STEEL_E单元的应力结果，存入变量STRSS_ST；用“*GET,STRSS_CO,ELEM,COPPER_E,ETAB,STRS_CO”从单元表STRS_CO中提取COPPER_E单元的应力结果，存入变量STRSS_CO。

(7) 将计算结果及其与理论解的比较情况输出到文件。

3. ANSYS程序

```

FINISH          ! 退出以前的模块
/CLEAR,START    ! 清除系统中的所有数据,读入启动文件的设置
! (1) 定义分析类型、单元类型、截面面积和材料的力学、热学参数
/FILNAME,EX1.4  ! 指定当前工程的文件名,所有数据用该文件名保存
/PREP7          ! 加载前处理模块
/TITLE, EX1.4, THERMALLY LOADED SUPPORT STRUCTURE
ANTYPE,STATIC    ! 定义分析类型为静力分析
ET,1,LINK1       ! 定义第1类单元为二维杆单元LINK1
R,1,65e-6        ! 定义第1类实常数, 杆件截面面积65mm^2
MP,EX,1,100E9    ! 定义第1类材料铜的弹性模量100 GPa
MP,ALPX,1,1.6E-5 ! 定义第1类材料铜的热膨胀系数1.6E-5
MP,EX,2,210E9    ! 定义第2类材料铁的弹性模量210 GPa
MP,ALPX,2,1.3E-5 ! 定义第2类材料铁的热膨胀系数1.3E-5
TREF,70          ! 设定参考温度为70度
! (2) 定义节点位置信息
N,1,-1           ! 在(-1m,0)处定义1号节点
N,3,1            ! 在(1m,0)处定义3号节点
FILL             ! 利用填充命令在1号和3号节点之间生成2号节点
N,4,-1,-1        ! 在(-1m,-1m)处定义4号节点
N,6,1,-1         ! 在(1m,-1m)处定义6号节点
FILL             ! 利用填充命令在4号和6号节点之间生成5号节点
! (3) 定义单元信息
E,1,4            ! 过节点1、4定义铜杆单元,使用默认的单元类型和材料类型
E,3,6            ! 过节点3、6定义铜杆单元
MAT,2           ! 改变材料类型为2
E,2,5            ! 过节点2、5定义铁杆单元
CP,1,UY,5,4,6    ! 定义5、4、6这3个节点的UY为耦合自由度,即3者的UY位移总是相等

```

```

! (4) 定义位移约束和荷载 (外力和温度荷载)
D,1,ALL,,3      ! 定义1号节点到3号节点全部固定
F,5,FY,-4000    ! 在5号节点处施加沿着-Y方向的大小等于4000N的集中力
BFUNIF,TEMP,80  ! 温度从原来的70度均匀升到80度 (TREF+10)
FINISH          ! 退出前处理模块PREP7
! (5) 在求解模块,定义求解选项,并开始求解
/SOLU           ! 加载求解模块
OUTPR,BASIC,1   ! 选择第1荷载步的基本输出项目
OUTPR,NLOAD,1   ! 指定输出第1荷载步的内容
NSUBST,1        ! 指定当前求解的荷载步
SOLVE           ! 开始求解
FINISH          ! 退出求解模块SOLU
! (6) 在一般后处理模块中,寻找特定位置的节点和单元,并提取它们的内力
/POST1          ! 加载后处理模块
STEEL_N = NODE (,,) ! 选择距离 (0,0,0) 位置最近的节点,保存到变量STEEL_N
COPPER_N = NODE (1,0,0) ! 选择距离 (1,0,0) 位置最近的节点,保存到变量COPPER_N
STEEL_E = ENEARN (STEEL_N) ! 选择距离STEEL_N节点最近的单元,保存到变量STEEL_E
COPPER_E = ENEARN (COPPER_N) ! 选择距离COPPER_N节点最近的单元,保存到变量COPPER_E
ETABLE,STRS_ST,LS,1 ! 用轴向应力SAXL的编号"LS,1"定义单元表STRS_ST
ETABLE,STRS_CO,LS,1 ! 用轴向应力SAXL的编号"LS,1"定义单元表STRS_CO
*GET,STRSS_ST,ELEM,STEEL_E,ETAB,STRS_ST ! 从单元表STRS_ST提取STEEL_E单元的
应力,存入STRSS_ST
*GET,STRSS_CO,ELEM,COPPER_E,ETAB,STRS_CO ! 从单元表STRS_CO提取COPPER_E单
元的应力,存入STRSS_CO
! (7) 将计算结果及其与理论解的比较情况输出到文件vm3.txt
*DIM,LABEL,CHAR,2,2 ! 定义2*2的字符型数组LABEL
*DIM,VALUE,,2,3      ! 定义2*3的数值型数组VALUE
LABEL(1,1) = 'STRSS_ST','STRSS_CO'
LABEL(1,2) = ' (MPa) ',' (MPa) '
*VFILL,VALUE(1,1),DATA,34.59,13.47
*VFILL,VALUE(1,2),DATA,STRSS_ST/1e6,STRSS_CO/1e6
*VFILL,VALUE(1,3),DATA,ABS(STRSS_ST/34.59e6),ABS(STRSS_CO/13.47e6)
/COM
/OUT,EX1_4,out ! 将输出内容重定向到文件EX1_4.out
/COM,----- EX1.4 RESULTS COMPARISON -----
/COM,
/COM,          |   TARGET   |   ANSYS   |   RATIO
/COM,
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,' ',F10.2,' ',F10.2,' ',1F5.3)
/COM,-----
/OUT          ! 结束数据重定向,关闭输出文件
FINISH        ! 退出后处理模块
*LIST,EX1_4,out ! 列表显示文件EX1_4.out的内容

```

4. 结果分析

经过以上的有限元计算，得到铜杆和铁杆中的轴向应力分别为：

$$N_c = 13.47 \text{ MPa}, N_s = 34.59 \text{ MPa}$$

与理论计算结果完全一致。

1.5 端部有间隙的杆的热膨胀

在两个相距3.002m的平行刚性壁之间，有一根铝合金杆一端固定在刚性壁上，在摄氏21度时杆件另一端距离刚性壁 $\delta = 2 \text{ mm}$ （如图1.14所示）。当温度升高到摄氏77度，试确定杆件内部的温度应力。弹性模量取 $E = 72.4 \text{ GPa}$ ，泊松比取0，热膨胀系数取 $\alpha = 2.25 \times 10^{-5} / ^\circ\text{C}$ 。

1. 理论解

在温度升高 $T = 77 - 21 = 56 ^\circ\text{C}$ 后，杆件的无约束膨胀伸长量为：

$$\Delta l = \alpha T l = 2.25 \times 10^{-5} \times 56 \times 3000 = 3.78 \text{ mm}$$

由于刚性壁的限制，实际伸长量为2mm。因此刚性壁给杆件的作用力使其缩短了1.78mm，因此可以确定出杆件的竖向应力为：

$$\sigma_y = 1.78 / 3000 \times E = 1.78 / 3000 \times 72.4 \times 10^3 = 42.957 \text{ MPa}$$

在固定端位置处，温度升高时杆件无法横向自由膨胀，水平方向应变保持为零，因此在水平方向产生温度应力。温度应力引起的应变等于热膨胀引起的应变，所以有：

$$\sigma_x = E \alpha T = 72.4 \times 10^3 \times 2.25 \times 10^{-5} \times 56 = 91.224 \text{ MPa}$$

2. Ansys解

利用平面单元，将杆件划分成如图1.15所示的有限元模型，由4个点构成一个长度为 l 的矩形区域，区域的下方留有一个距离刚性壁3mm的间隙。刚性壁由接触单元的11，12号节点构成。本例中使用了平面单元PLANE42，它是一个4节点的平面单元。接触单元使用了点对面的二维接触单元CONTAC26，它约束了上面平面单元上的节点1，节点2穿过刚性平面。




图1.14 端部有间隙的杆的热膨胀




图1.15 节点布置和单元划分

下面程序首先定义了本问题中的两类单元的一些力学参数，然后定义了构成平面区域的4个节点和下面刚性壁的两个节点。单元定义则只有两个单元，但是它们具有不同的单元

类型，所以在定义第2个单元时需要将默认的第1类平面单元类型改编为第2类接触单元类型，材料类型也改为第2种接触面材料。

对3, 4号节点的位移约束取为完全固定, 1, 2节点则只限制水平方向的位移自由度。单元表的定义指定了水平应力和铅直应力, 很明确地指定要提取3号单元的水平应力和铅直应力。经过下面程序中的前4步, 得到如图1.16所示的定义了相应边界条件的杆件模型。




图1.16 端部有间隙的杆件的接触问题有限元模型

下面是全部代码，注意：这里计算单位取的是N和mm，应力单位是MPa。

3. ANSYS程序

```
FINISH          ! 退出以前的模块
/CLEAR,START    ! 清除系统中的所有数据,读入启动文件的设置
! (1) 定义分析类型、单元类型、平面单元厚度和材料的力学、热学参数
/FILNAME, EX1.5  ! 指定当前工程的文件名,所有数据用该文件名保存
/PREP7          ! 加载前处理模块
/TITLE, EX1.5, THERMAL EXPANSION TO CLOSE A GAP AT A RIGID SURFACE
ANTYPE,STATIC    ! 定义分析类型为静力分析
ET,1,CONTAC26    ! 定义第1类单元为二维接触单元CONTACT26
R,1,2.625E15     ! 定义接触单元的接触刚度 K = 2.625E15
ET,2,PLANE42,,,3 ! 定义第2类单元为二维带厚度的平面应力单元PLANE42
R,2,1           ! 定义第2类实常数,平面单元的厚度1mm
MP,EX,1,72.4E3  ! 定义第1类材料的弹性模量EX = 72.4GPa
MP,ALPX,1,2.25E-5 ! 定义第1类材料的热膨胀系数ALPX = 2.25E-5
MP,NUXY,1,0     ! 定义第1类材料的泊松比 NUXY = 0
! (2) 定义节点信息
N,1,2000,1000   ! 在 (2000mm,1000 mm) 处定义1号节点
N,2,3000,1000   ! 在 (3000 mm,1000 mm) 处定义2号节点
```

```

N,3,3000,4000      ! 在 (3000 mm,4000 mm) 处定义3号节点
N,4,2000,4000      ! 在 (2000 mm,4000 mm) 处定义4号节点
N,11,1000,998      ! 在 (1000 mm,998 mm) 处定义11号节点
N,12,4000,998      ! 在 (4000 mm,998 mm) 处定义12号节点
! (3) 定义单元信息
E,1,11,12          ! 过11、12号节点定义接触单元
E,2,11,12
TYPE,2             ! 选择单元类型为2
REAL,2             ! 选择实常数类型为2
E,1,2,3,4          ! 过节点1、2、3、4定义平面单元
TREF,21            ! 设定起始参考温度21度
BFUNIF,TEMP,77     ! 将所有节点的温度均匀提升到77度
! (4) 定义位移约束
D,3,ALL,,4         ! 约束3号节点的所有位移,并循环到4号节点
D,1,UX,,2,1        ! 约束1号节点的UX位移,并循环到2号节点
OUTPR,BASIC,LAST   ! 选择基本输出选项,直到最后一个荷载步
FINISH             ! 退出前处理模块PREP7
! (5) 在求解模块开始求解
/SOLU              ! 加载求解模块
SOLVE              ! 开始求解
FINISH             ! 退出求解模块SOLU
! (6) 在一般后处理模块中,定义水平应力和铅直应力单元表,并提取3号单元的应力结果
/POST1             ! 进入一般后处理模块
ETABLE,STRSX,S,X    ! 定义X方向的应力为单元表STRSX
ETABLE,STRSY,S,Y    ! 定义Y方向的应力为单元表STRSY
*GET,STRSSX,ELEM,3,ETAB,STRSX ! 从单元表STRSX提取3号单元X方向的应力,存入STRSSX
*GET,STRSSY,ELEM,3,ETAB,STRSY ! 从单元表STRSY提取3号单元Y方向的应力,存入STRSSY
! (7) 将计算结果输出到结果文件
*DIM,LABEL,CHAR,2,2 ! 定义2*24个元素的字符型数组LABEL
*DIM,VALUE,,2,3     ! 定义2*3的数值型数组VALUE
LABEL(1,1) = 'SIGX, (M', 'SIGY, (M'
LABEL(1,2) = 'Pa)    ', 'Pa)    '
*VFILL,VALUE(1,1),DATA,-91.22,-42.96
*VFILL,VALUE(1,2),DATA,STRSSX,STRSSY
*VFILL,VALUE(1,3),DATA,ABS(STRSSX/91.22),ABS(STRSSY/42.96)
/COM
/OUT,EX1_5, out     ! 将输出内容重定向到计算结果文件
/COM,----- EX1.5 RESULTS COMPARISON -----
/COM,
/COM,              |  TARGET  |  ANSYS   |  RATIO
/COM,
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)

```

```
(1X,A8,A8,' ',F10.3,' ',F10.3,' ',1F5.3)
/COM,-----
/OUT      ! 结束数据重定向,关闭输出文件
FINISH    ! 退出后处理模块
*LIST,EX1_5,out ! 列表显示计算结果文件的内容
```

4. 结果分析

经过以上有限元接触分析,程序很快收敛,得到3号单元中的水平最大温度应力为 $\sigma_x=91.224\text{MPa}$,最大竖向应力 $\sigma_y=42.957\text{MPa}$,与前面的理论计算结果完全吻合。

第7章 简单振动系统

7.1 单自由度弹簧质量系统的频率计算

有一个重量 $w=11.121\text{N}$ 的仪器放置在刚度为 $k=8406.4\text{N/m}$ 的橡胶支座上，试确定该仪器自由振动的频率。




图7.1 单自由度弹簧振子有限元模型




图7.2 利用直接建模生成的有限元模型

1. 理论解

按照文献[9]第6页码的练习1.2-2，根据单自由度弹簧质量系统的频率公式，可确定出该系统的自由振动频率为：

$$f = \frac{1}{2\pi} \sqrt{\frac{k}{m}} = \frac{1}{2 \times 3.14159} \sqrt{\frac{8406.4}{11.121/9.8}} = 13.7\text{Hz}$$

2. ANSYS求解

这是一个模态分析问题。在ANSYS中，用于振动系统模态计算的单元主要有：弹簧阻尼（COMBINE）单元和质量单元（MASS）。本问题中的集中质量只有一个自由度，所以系统的振动模态和振动频率也只有一个。定义模型只需要两个节点就够了，一个节点是集中质量单元，另一个是固定点约束，中间连接的单元是弹簧阻尼单元。由于有两类单元出现在本问题中，所以单元类型定义和实常数的定义，显得比模型本身定义更复杂一点。

首先，需要设置模型求解的许多选项，包括文件名称FILENAME（用于自动保存许多中间数据文件），标题名称TITLE（主要出现在图形窗口上），分析类型ANTYPE为模态分析（MODAL或者取编号2）及其模态分析选项等。用ET（Element Type）定义需要用到

的两类单元：COMBIN14和MASS21，以及它们的实常数（Real constant）。

然后，利用节点定义命令N（Node）定义节点位置，用单元定义命令E（Element）定义单元连接关系。这里的第1号弹簧单元由节点1和节点2组成，第2号集中质量单元只有一个2号节点。需要注意的是，在定义这个单元时需要用单元类型TYPE和实常数类型REAL命令切换单元类型和实常数类型。为了演示如何减小计算自由度数目，这里使用了定义主自由度命令M（Master DOF）将2号节点的UY方向设为主自由度。

最后，对1节点施加了位移固定约束，2号节点UX方向的自由度也被约束后，利用求解模块（/SOLUTION）中的求解命令“SOLVE”开始求解。求解后的频率结果，可以在一般后处理模块/POST1中，利用获取变量值的命令“*GET”取出，存入变量FREQ。读者也可以通过GUI界面上的“Utility Menu->Parameters->Scale Parameters...”打开变量定义显示窗口，看到该变量的值。也可以利用“Utility Menu->List->Other->Parameters”菜单，列表显示出所有变量的值。

下面给出的ANSYS程序可使用“Utility Menu->File->Read Input From...”菜单加载执行后，生成计算结果的文本文件“EX7_1.out”，该文本文件可以用任何文本编辑器打开。

注意：下面程序中使用的第（6）段内容，只有在“Read Input From...”方式下有效。如果读者是用拷贝（CTRL + C），粘贴（CTRL + V）到ANSYS命令行的方式执行，只需要将前面5段程序粘贴过来，就可以看出执行过程和结果了。

下面给出该问题分析的主要步骤和命令说明。

（1）设置工程选项、求解选项、单元类型和实常数。

用工程文件名称定义命令“/FILENAME, EX7.1”定义当前工程文件名称为EX7.1，在以下的计算过程中，所有的中间文件都用EX7.1名称命名，只是各类不同的文件具有不同的文件名后缀。用加载前处理模块命令“/PREP7”进入前处理模块，在前处理模块，所有用于前处理操作的命令才可以被识别和处理，而其他模块的命令是不能在这里使用的。用标题定义命令“/TITLE”定义标题信息，标题信息主要出现在图形窗口底部。

用分析类型定义命令“ANTYPE, MODAL”定义当前工程的分析类型为模态（MODAL）分析，并用模态分析选项定义命令“MODEXT, REDUC, 1,, 1”定义显示所有模态，用模态展开个数定义命令“MXPAND, 1”指定只展开第1阶模态。

用单元类型定义命令“ET, 1, COMBIN14,, 2”定义第1类单元为2D纵向弹簧单元COMBIN14，用“ET, 2, MASS21,, 4”定义第2类单元为2D质量单元MASS21。

用实常数定义命令“R, 1, 8406.4”定义第1类实常数，弹簧刚度系数 $k=8406.4$ N/m。用“R, 2, 11.121/9.8”定义第2类实常数，集中质量 $m=11.121/9.8$ kg。这里使用了表达式赋值，在ANSYS中的许多地方都支持表达式赋值。

（2）定义模型节点位置和单元连接关系。

用节点定义命令“N, 1”在坐标原点（0,0,0）定义第1号结点，这里的“0”可以被省略。用“N, 2, 1”在坐标（0,1m,0）定义第2号结点。

用单元定义命令“E, 1, 2”过1号节点和2号节点定义单元。用当前单元类型定义命令“TYPE, 2”将默认的单元类型从第1类弹簧单元，改变为第2类质量单元。在使用这个命令之前，ANSYS默认为单元类型为1，实常数类型也是1。用当前实常数类型定义命令“REAL, 2”

将默认的实常数从第1类实常数（弹簧刚度），改变为第2类实常数（集中质量）。这两个命令用于改变以下定义单元时所使用的单元类型和实常数，也就是说，前面已经定义的单元使用的是第1类单元和第1类实常数，而下面要定义的单元则要使用第2类单元和第2类实常数。

单元定义命令“E,2”用第2个节点定义第2个单元。主自由度定义命令“M,2,UY”指定了2号节点的主自由度为UY方向。

(3) 指定输出选项、位移约束。

用计算结果输出选项定义命令“OUTPR,ALL,1”指定只输出第1阶模态的所有内容。用“OUTRES,ALL,0”设置将所有数据不记录到数据库。

用位移约束定义命令“D,1,ALL”约束1号节点的所有位移自由度，用“D,2,UX”约束2号节点的UX方向自由度。利用步骤(1)到(3)，生成了如图7.2所示的有限元模型。

(4) 进入求解模块并求解。

(5) 提取计算结果。

用计算结果提取命令“*GET,FREQ,MODE,1,FREQ”提取第1阶模态的频率计算结果，并保存到变量FREQ中。通过GUI界面上的“Utility Menu -> Parameters -> Scale Parameters...”打开变量定义显示窗口，看到该变量的值。

(6) 定义数组变量，将计算结果输出到文本文件。

3. ANSYS源程序

```
FINISH          ! 退出以前模块
/CLEAR, NOSTART ! 清除数据，不加载初始化文件
!(1) 设置工程选项、求解选项、单元类型和实常数
/FILENAME, EX7.1 ! 定义工程文件名称
/PREP7          ! 加载前处理模块
/TITLE, EX7.1, NATURAL FREQUENCY OF A SPRING-MASS SYSTEM
C*** VIBRATION THEORY AND APPLICATIONS, THOMSON, 2ND, PAGE 6, EX.1.2-2
ANTYPE,MODAL    ! 定义分析类型为模态(MODAL)分析
MODEOPT,REDUC,1,,,1 ! 显示所有模态
MXPAND,1        ! 展开第1阶频率
ET,1,COMBIN14,,,2 ! 定义第1类单元为2D纵向弹簧单元COMBIN14
ET,2,MASS21,,,4 ! 定义第2类单元为2D质量单元MASS21
R,1,8406.4       ! 定义第1类实常数,弹簧刚度系数k=8406.4 N/m
R,2,11.121/9.8   ! 定义第2类实常数,集中质量m=11.121/9.8 kg
!(2) 定义模型节点和单元连接关系
N,1              ! 在坐标原点(0,0,0)定义第1号结点
N,2,,1          ! 在坐标(0,1,0)定义第2号结点
E,1,2           ! 过1号节点和2号节点定义单元
TYPE,2          ! 将默认的单元类型从弹簧单元,改变为第2类单元(质量单元)
REAL,2          ! 将默认的实常数从弹簧刚度,改变为第2类实常数(集中质量)
E,2             ! 用第2个节点定义第2个单元
M,2,UY          ! 指定2号节点的主自由度为主自由度
```

```

!(3) 指定输出选项,位移约束
OUTPR,ALL,1    ! 输出第1阶频率的所有内容
OUTRES,ALL,0   ! 所有数据不记录到数据库
D,1,ALL        ! 约束1号节点的所有位移自由度
D,2,UX         ! 约束2号节点的X方向自由度UX
FINISH        ! 退出前处理模块PREP7
!(4) 进入求解模块求解
/SOLU          ! 加载求解模块
SOLVE          ! 开始求解
!(5) 提取计算结果
*GET,FREQ,MODE,1,FREQ ! 提取第1阶模态的频率计算结果,存入变量FREQ
!(6) 定义数组变量,将计算结果输出到文本文件
*DIM,LABEL,CHAR,1,2 ! 定义1*2的字符型变量LABEL
*DIM,VALUE,,1,3      ! 定义默认为数值型的1*3的数值型变量VALUE
LABEL(1,1) = '      F,' ! 给数组元素LABEL(1,1)变量赋值字符串
LABEL(1,2) = '(Hz)  '   ! 给数组元素LABEL(1,2)变量赋值字符串
*VFILL,VALUE(1,1),DATA,13.701 ! 给数组元素LABEL(1,1)变量赋理论值13.701
*VFILL,VALUE(1,2),DATA,FREQ   ! 给数组元素LABEL(1,2)变量赋计算值FREQ
*VFILL,VALUE(1,3),DATA,ABS(FREQ/13.701) ! 给数组元素LABEL(1,3)变量赋频率计算结果和理论值的比值
/COM
/OUT, EX7_1, out ! 将输出内容重定向到文件EX7_1.out
/COM,----- EX7.1 RESULTS COMPARISON -----
/COM,
/COM,          |  TARGET  |  ANSYS  |  RATIO
/COM,
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,'    ',F10.3,'    ',F10.3,'    ',1F5.3)
/COM,-----
/OUT          ! 结束数据重定向,关闭输出文件
FINISH        ! 退出求解模块SOLU
*LIST, EX7_1, out ! 列表显示结果文件EX7_1.out的内容

```

4. 结果比较和讨论

通过上面程序的计算,得到了如表7.1所示的计算结果,表中的理论值是前面的理论计算结果,ANSYS值是程序计算结果。通过比较可以看出,尽管选取了2个节点一个自由度,计算结果和理论结果非常接近,说明ANSYS有限元分析系统的模态分析功能对刚体振动系统,仍然具有非常精确的计算能力。

表7.1 单自由度振动系统的有限元计算结果和理论结果的比较

频率阶数	理论值	ANSYS值	比值
第1阶频率(Hz)	13.701	13.698	1.000

7.2 悬索自由振动的频率

长度 $l=2.54\text{m}$ 的柔软绳索，张紧后用铰链固定连接在两刚性支座之间（如图7.3所示），已知张紧后的轴向应变为 $\varepsilon_0=0.543228\%$ ，材料密度 $\rho=7727.4\text{kg/m}^3$ ，弹性模量 $E=207\text{GPa}$ ，横截面积 $A=1.97925\text{mm}^2$ ，试利用三维拉压杆单元LINK10确定在该状态下绳索内的拉应力和张力，以及纵向自由振动的前3阶频率。




图7.3 悬索纵向自由振动分析的有限元模型

1. 理论解

按照文献[9]的264页，练习8.2中的公式。根据拉压杆的虎克定律，在轴向应变 ε_0 时，绳索内的应力和张力分别为：

$$\sigma = E \varepsilon_0 = 207 \times 10^9 \times 0.00543228 = 1124.5 \text{ MPa}$$

$$T = A \sigma = AE \varepsilon_0 = 1.97925 \times 10^{-6} \times 207 \times 10^9 \times 0.00543228 = 2225.63 \text{ N}$$

单位长度为 ρA 的柔软绳索在张力 T 作用下自由振动的方程为：

$$\frac{\partial^2 y(x,t)}{\partial x^2} = \frac{1}{c^2} \frac{\partial^2 y(x,t)}{\partial t^2}$$

其中， $c = \sqrt{T/(\rho A)}$ 是波沿着绳索纵向传播的速度。

假设波动解满足分离变量形式 $y(x,t) = Y(x)G(t)$ ，代入到前面的振动方程后得到：

$$\frac{1}{Y(x)} \frac{d^2 Y(x)}{dx^2} = \frac{1}{c^2} \frac{1}{G(t)} \frac{d^2 G(t)}{dt^2}$$

$$\frac{1}{Y(x)} = \frac{d^2 Y(x)}{dx^2} = \frac{1}{c^2} \frac{1}{G(t)} \frac{d^2 G(t)}{dt^2}$$

上面方程的左侧只与 x 有关，与 t 无关；方程的右侧只与 t 有关，与 x 无关。所以两端就必须等于一个常数，设这个常数为 $-(\omega/c)^2$ 。这样就得到了两个常微分方程：

$$\frac{d^2 Y(x)}{dx^2} + \left(\frac{\omega}{c}\right)^2 Y(x) = 0$$

$$\frac{d^2 G(t)}{dt^2} + \omega^2 G(t) = 0$$

这两个常微分方程的通解为：

$$Y(x) = A \sin \frac{\omega}{c} x + B \cos \frac{\omega}{c} x$$

$$G(t) = C \sin \omega t + D \cos \omega t$$

其中的积分常数A, B由边界条件确定, C, D由初始条件确定。

对于两端固定的张紧绳索, 在 $x=0$ 和 $x=l$ 处有位移边界条件:

$$y(0,t)=y(l,t)$$

由 $x=0$ 处的条件确定出 $B=0$ 。此时振动解变为:

$$y(x,t)=(C\sin \omega t + D\cos \omega t)\sin \frac{\omega}{c}x$$

另一端 $x=l$ 处的边界条件 $y(l,t)=0$ 确定出:

$$\sin \frac{\omega l}{c}=0$$

即

$$\frac{\omega_n l}{c} = \frac{2\pi l}{\lambda} = n\pi, \quad (n=1,2,3,\dots)$$

其中 $\lambda=c/f$ 是波长, f 是振动频率。因此我们得到柔软绳索纵向自由振动的频率计算公式为:

$$f_n = \frac{n}{2l}c = \frac{2n}{2l}\sqrt{\frac{T}{\rho A}}, \quad (n=1,2,3,\dots)$$

将已知数据代入上面公式, 得到:

$$f_1 = \frac{1}{2l}\sqrt{\frac{T}{\rho A}} = \frac{1}{2 \times 2.54}\sqrt{\frac{2225.63}{7727.4 \times 1.97925 \times 10^{-6}}} = 75.09\text{Hz}$$

$$f_2 = \frac{2}{2l}\sqrt{\frac{T}{\rho A}} = \frac{2}{2.54}\sqrt{\frac{2225.63}{7727.4 \times 1.97925 \times 10^{-6}}} = 150.18\text{Hz}$$

$$f_3 = \frac{3}{2l}\sqrt{\frac{T}{\rho A}} = \frac{3}{2 \times 2.54}\sqrt{\frac{2225.63}{7727.4 \times 1.97925 \times 10^{-6}}} = 225.27\text{Hz}$$

2. ANSYS求解

在长度2.54m的范围内布置14个节点, 划分13个单元(如图7.3所示), 固定第1个节点和最后一个13号节点。由于结构在施加荷载之前已经有初始的张力, 这里使用预应力开关来模拟这种状态。

频率计算使用子空间迭代方法, 其他求解频率的方法也可以使用。这里计算前3阶频率。下面给出的是分析程序的主要步骤和说明。

(1) 进入前处理模块/PREP7, 定义工程选项、分析类型、单元类型、实常数和材料参数。

用ANTYPE申明为静力分析, 先计算有预应力情况下的拉应力和张力。用ET命令定义单元类型为二维杆单元LINK1, 并用实常数定义命令R定义杆件的横截面面积为1.97925平方毫米, 初应变0.54322%。用材料定义命令MP定义弹性模量EX = 209GPa, 密度DENS = 7727.4kg/m³。

(2) 定义节点和单元。

用节点定义命令N在坐标原点(0,0,0)定义结点1, 在坐标(2.54m,0,0)定义结点14。

使用节点填充命令FILL在结点1和结点14中间填充生成其他节点。

用单元定义命令E过节点1, 2定义第一个单元, 用单元生成命令EGEN (Element GENeration) 按照第一个单元的模式循环13次生成其他12个单元, 每次单元两端的节点号码增加1。利用以上命令生成了如图7.4所示的有限元模型。




图7.4 利用节点和单元直接建模生成的有限元模型

(3) 设置输出选项和位移约束。

用“OUTPR,BASIC,1”指定只输出第1阶模态的基本计算结果。用“D,ALL,ALL”固定所有节点的所有自由度, 以进行有预应力的静力分析。施加这些位移约束后, 模型变成如图7.5所示的样子。




图7.5 施加位移约束后的模型

(4) 在求解模块/SOLU中, 求解预应力引起的应力状态。

(5) 在一般后处理模块/POST1, 定义单元表, 提取单元的应力计算结果。

用单元表定义命令ETABLE以轴向应力“LS,1”为内容, 定义单元表STRS。用获取单元表数据命令“*GET”从单元表STRS中提取13号单元的轴向应力, 存入变量STRSS。

(6) 在时间历程后处理模块/POST26中, 计算支座反力。

用反力计算命令“RFORCE,2,1,F,X”计算1号节点在X方向的支座反力, 并存入POST26模块中的第2个变量。用STORE命令保存数据到数据库, 用“*GET”命令从数据库中提取第2个变量的值, 存入变量FORCE。这些变量的值可以通过GUI界面上的“Utility Menu -> Parameters -> Scale Parameters ...”打开变量定义显示窗口, 看到该变量的值(如图7.6所示)。

(7) 再次进入求解模块/SOLU, 进行模态分析。

用分析类型定义命令“ANTYPE,MODAL”定义分析类型为模态分析, 并用模态分析选项定义命令“MODOPT,SUBSP,3”指定使用子空间循环迭代方法展开前3个模态, 设置“PSTRES,ON”选项包括预应力效应。用删除位移约束命令“DDELE,2,UX,13”释放2号节点到13号节点的所有X方向的自由度UX, 用“DDELE,2,UY,13”释放2号节点到13号节点的所有Y方向的自由度UY。

(8) 获取前面3阶频率

用“*GET,FREQ1,MODE,1,FREQ”获取第1阶频率, 存入变量FREQ1, 同样可以提取出第2阶和第3阶频率, 分别保存到变量FREQ2和FREQ3。这些变量的值可以通过GUI界面上的“Utility Menu -> Parameters -> Scale Parameters ...”打开变量定义显示窗口, 看到该变量的值(如图7.7)。



图7.6 利用*GET命令提取出的静力计算结果




图7.7 用*GET提取出的模态分析计算结果

(9) 定义数组变量, 将计算结果存入文本文件EX7_2.out。

3. ANSYS源程序

```
FINISH
/CLEAR, NOSTART      ! 清除数据, 不加载初始化文件
!(1) 定义工程选项、分析类型单元类型、实常数和材料参数
```



```

/FILENAME, EX7.2      ! 定义工程文件名
/PREP7      ! 加载前处理模块
/TITLE, EX7.2, VIBRATION OF A STRING UNDER TENSION
C***  VIBRATION THEORY AND APPLICATIONS, THOMSON, 2ND., PAGE 264, ART. 8.2,
ANTYPE, STATIC      ! 定义分析类型为静力分析
PSTRES, ON          ! 打开预应力开关(该选项对于有预应力的振动分析是非常重要的)
ET, 1, LINK1        ! 定义第1类单元为二维杆单元LINK1
R, 1, 1.97925E-6, 0.54322E-2 ! 定义杆的截面积1.97925mm^2和初应变0.54322%
MP, EX, 1, 207E9      ! 定义第1类材料的弹性模量为EX=207GPa
MP, DENS, 1, 7727.4    ! 定义第1类材料的密度DENS=7727.4kg
!(2) 定义节点和单元
N, 1                ! 在坐标原点(0,0,0)定义结点1
N, 14, 2.54         ! 在坐标(2.54m,0,0)定义结点14
FILL                ! 在结点1和结点14中间填充生成其他节点
E, 1, 2             ! 过节点1,2定义单元
EGEN, 13, 1, 1      ! 按照前面定义的单元模式循环13次生成其他12个单元,每次节点号增加1
!(3) 设置输出选项和位移约束
OUTPR, BASIC, 1      ! 只输出第1阶模态的基本计算结果
D, ALL, ALL          ! 固定所有节点,进行有预应力的静力分析
FINISH              ! 退出前处理模块PREP7
!(4) 在求解模块,求解预应力引起的应力状态
/SOLU                ! 加载求解模块
SOLVE                ! 开始求解
FINISH              ! 退出求解模块SOLU
!(5) 在后处理模块,定义单元表,提取单元的应力计算结果
/POST1              ! 进入一般后处理模块
ETABLE, STRS, LS, 1   ! 以轴向应力"LS,1"为内容,定义单元表STRS
*GET, STRSS, ELEM, 13, ETAB, STRS ! 从单元表STRS中提取13号单元的轴向应力,存入变量STRSS
FINISH              ! 退出后处理模块
!(6) 在时间历程后处理模块中,计算支座反力
/POST26             ! 进入时间历程后处理器Post26
RFORCE, 2, 1, F, X    ! 计算1号节点的x方向的支座反力,存入第2个变量
STORE               ! 保存数据
*GET, FORCE, VARI, 2, EXTREM, VMAX ! 从数据库中提取第2个变量的值,存入变量FORCE
!(7) 再次进入求解模块,进行模态分析
/SOLU                ! 加载求解模块
ANTYPE, MODAL        ! 定义分析类型为模态分析
MODOPT, SUBSP, 3      ! 用子空间循环迭代方法计算前3个模态
MXPAND, 3            ! 展开前3阶模态
PSTRES, ON           ! 包括预应力效应
DDELE, 2, UX, 13      ! 释放2号节点到13号节点的所有x方向的自由度UX
DDELE, 2, UY, 13      ! 释放2号节点到13号节点的所有y方向的自由度UY
SOLVE                ! 开始模态分析的求解
!(8) 获取前面3阶频率

```

```
*GET,FREQ1,MODE,1,FREQ      ! 获取第1阶频率,存入变量FREQ1
*GET,FREQ2,MODE,2,FREQ      ! 获取第2阶频率,存入变量FREQ2
*GET,FREQ3,MODE,3,FREQ      ! 获取第3阶频率,存入变量FREQ3
!(9) 定义数组变量,将计算结果存入文件
*DIM,LABEL,CHAR,5,2
*DIM,VALUE,,5,3
LABEL(1,1) = '      F','      SI','      f1','      f2','      f3'
LABEL(1,2) = ' N      ','GMA,MPa ','      Hz      ','      Hz      ','      Hz      '
*VFILL,VALUE(1,1),DATA,2225.63, 1124.5,75.09,150.18,225.27
*VFILL,VALUE(1,2),DATA,ABS(FORCE),STRSS/1e6,FREQ1,FREQ2,FREQ3
*VFILL,VALUE(1,3),DATA,ABS(FORCE/2225.63),ABS(STRSS/1124.5e6),ABS(FREQ1/75.09)
*VFILL,VALUE(4,3),DATA,ABS(FREQ2/150.18),ABS(FREQ3/225.27)
/COM
/OUT, EX7_2, out      ! 将输出内容重定向到文件
/COM,----- EX7.2 RESULTS COMPARISON -----
/COM,
/COM,          |  TARGET  |  ANSYS  |  RATIO
/COM,
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,'      ',F10.3,'      ',F10.3,'      ',1F5.3)
/COM,-----
/OUT      ! 结束数据重定向,关闭输出文件
FINISH      ! 退出后处理模块
*LIST, EX7_2, out      ! 列表显示文件内容
```

4 . 结果比较和讨论

利用上面程序计算得到表7.2所示的计算结果及其与理论结果的比较。静力分析结果与理论分析结果吻合得相当好。模态分析的误差随着频率阶次增加而增大，但是如果加密单元后，情况会大大改善，读者可以增加节点，分析各阶频率的变化情况。

表7.2 有预应力的绳索的应力计算和纵向自由振动频率

预应力和频率阶数	理论解	Ansys解	比值
张力F(N)	2225.630	2225.598	1.000
轴向应力(Mpa)	1124.500	1124.465	1.000
第1阶频率f1(Hz)	75.090	75.480	1.005
第2阶频率f2(Hz)	150.180	152.063	1.013
第3阶频率f3(Hz)	225.270	230.864	1.025

7.3 用弹簧单元连接的圆盘的扭转振动

转动惯量 $J=0.034235\text{kg} \cdot \text{m}^2$ 的圆盘，固定在一根扭转刚度 $k=0.542337 \text{ N} \cdot \text{m/rad}$ 的杆件

的自由端，该杆件另一端固定（如图7.8所示）。试利用弹簧阻尼单元COMBINE14和集中质量单元MASS21确定该圆盘扭转自由振动的第1阶频率。




图7.8 扭转振动圆盘模型



图7.9 用弹簧元和质量元模拟圆盘的扭转振动

1. 理论解

按照文献[9]第10页上的练习1.3，圆盘扭转振动自由振动的频率公式为：

$$f_1 = \frac{1}{2\pi} \sqrt{\frac{k}{J}} = \frac{1}{2 \times 3.14159} \sqrt{\frac{0.542337}{0.034235}} = 0.63346 \text{ Hz}$$

2. ANSYS计算

由于这里使用弹簧阻尼单元COMBINE14和集中质量单元MASS21来模拟该振动系统，所以受扭杆的长度可以是任意的。为了消除圆盘其他运动，约束了除绕Z轴转动之外的所有自由度，并使用了主自由度的概念。将受扭杆件定义为弹簧单元，在其两端设置两个节点，在振动圆盘位置设置一个考虑转动惯量的集中质量单元，并将其转动自由度设置为主自由度（如图7.9所示）。下面是完成这个问题分析的主要步骤和说明。

(1) 设置工程选项、求解选项、单元类型和实常数。

用工程文件命名命令“/FileName, EX7.3”指定工程名称为EX7.3。进入前处理模块/REP7，用标题定义命令“/TITLE”定义工程标题，该标题信息会出现在图形窗口上。用分析类型定义命令“ANTYPE,MODAL”定义分析类型为模态分析（MODAL）。用模态分析选项命令“MODEOPT”定义使用降阶方法，显示第1阶模态，用频率展开数定义命令“MXPAND,1”展开第1阶频率。

用单元类型定义命令“ET,1,COMBIN14,,,1”定义第1类单元为三维考虑扭转效应的弹簧阻尼单元COMBIN14。用“ET,2,MASS21,,,3”定义第2类单元为带转动效应的二维质量单元MASS21。

用实常数定义命令“R,1, 0.542337”定义弹簧扭转刚度系数 $k=0.542337 \text{ N}\cdot\text{m}/\text{rad}$ ，用“R,2,1, 0.034235”定义质量单元的集中质量为1，转动惯量 $J=0.034235 \text{ kg}\cdot\text{m}^2$ 。

(2) 定义模型节点和单元连接关系。

为了便于观察模型建立过程，用视点定义命令“/VIEW,1,1,1,1”将视点切换到(1,1,1)等轴测位置。用节点定义命令“N,1”在坐标原点(0,0,0)定义第1个结点，这里所有0值都被省略了。用“N,2,,, -0.0254”在(0,0, -0.0254m)位置定义第2个结点。

用单元定义命令“E,1,2”过1,2结点定义第1个单元，用当前单元类型定义命令“TYPE,2”将默认的单元类型从第1类型弹簧单元COMBINE14，改变为第2类集中质量单元MASS21。用实常数类型改变命令“REAL,2”将默认的实常数类型从第1类实常数（弹簧刚度），改变为第2类实常数（集中质量）。用单元定义命令“E,2”在第2个节点处定义第2个单元，用主自由度定义命令“M,2,ROTZ”指定2号节点的主自由度为绕X轴的转动自由度ROTZ。

(3) 指定输出选项和位移约束。

用输出项定义命令“OUTPR,BASIC,1”设置只输出第1阶频率的基本内容。用位移约束定义命令“D,1,ALL”约束端点处的第1号节点的所有位移自由度。用“D,2,UX,,,,UY”限制2号节点的平移自由度UX和UY。其实这一句可以不要，因为连接2号节点的弹簧单元是只有扭转自由度的单元，2号节点不会发生平移。但是为了便于阅读和理解，也可以加上这一句。

利用前面步骤建立的模型，在施加位移约束，定义主自由度后，变成如图7.10所示的样子。1号节点处的标志是固定所有位移约束的标志，2号节点处的标志是主自由度标志。




图7.10 扭转振动圆盘的有限元模型和施加的位移约束，主自由度

(4) 进入求解模块 / SOLUT求解。

(5) 提取计算结果。

用提取计算结果命令“*GET,FREQ,MODE,1,FREQ”提取第1阶模态的频率计算结果，存入变量FREQ。提取出的频率结果如图7.11所示。

(6) 定义数组变量，将计算结果输出到文本文件。



图7.11 扭转圆盘的扭转自由振动的频率计算结果

3. ANSYS源程序

```

FINISH
/CLEAR, NOSTART      ! 清除数据,不加载初始化文件
!(1) 设置工程选项、求解选项、单元类型和实常数
/FileName, EX7.3      ! 定义工程文件名称
/PREP7               ! 加载前处理模块
/TITLE, EX7.3, TORSIONAL FREQUENCY OF A SUSPENDED DISK
C*** VIBRATION THEORY AND APPLICATIONS, THOMSON, 2ND, PAGE 10, EX 1.3-2
ANTYPE,MODAL          ! 定义分析类型为模态(MODAL)分析
MODEOPT,REDUC,,,1     ! 用降阶方法求解,显示第1阶振型
MXPAND,1              ! 展开第1阶频率
ET,1,COMBIN14,,,1     ! 定义第1类单元为考虑扭转的弹簧COMBIN14
ET,2,MASS21,,,3       ! 定义第2类单元为带转动惯量的二维质量单元MASS21
R,1,0.542337          ! 定义弹簧扭转刚度系数k=0.542337 N.m/rad
R,2,1,0.034235        ! 定义集中质量为m=1,转动惯量J=0.034235 kg.m^2
!(2) 定义模型节点和单元连接关系
/VIEW,1,1,1,1        ! 将视点切换到(1,1,1)等轴测位置
N,1                  ! 在坐标原点定义第1个结点
N,2,,, -0.0254       ! 在(0,0,-0.0254m)位置定义第2个结点
E,1,2                ! 过1,2结点定义单元
TYPE,2               ! 将默认的单元类型从弹簧单元,改变为第2类单元(质量单元)
REAL,2               ! 将默认的实常数从弹簧刚度,改变为第2类实常数(集中质量)
E,2                  ! 用第2个节点定义第2个单元
M,2,ROTZ             ! 指定2号节点的主自由度为绕x轴的转动自由度ROTZ
!(3) 指定输出选项和位移约束
OUTPR,BASIC,1        ! 输出第1阶频率的基本内容
D,1,ALL              ! 约束端点处的第1号节点的所有位移自由度
D,2,UX,,,,,UY        ! 约束2号节点的平移自由度
FINISH               ! 退出前处理模块PREP7

```

```

!(4) 进入求解模块求解
/SOLU          ! 加载求解模块
SOLVE          ! 开始求解
!(5) 提取计算结果
*GET,FREQ,MODE,1,FREQ ! 提取第1阶模态的频率计算结果,存入变量FREQ
!(6) 定义数组变量,将计算结果输出到文本文件
*DIM,LABEL,CHAR,1,2
*DIM,VALUE,,1,3
LABEL(1,1) = '      F,'
LABEL(1,2) = '(Hz)  '
*VFILL,VALUE(1,1),DATA,0.63346
*VFILL,VALUE(1,2),DATA,FREQ
*VFILL,VALUE(1,3),DATA,ABS(FREQ/0.63346 )
/COM
/OUT, EX7_3, out ! 将输出内容重定向到文件EX7_3.out
/COM,----- EX7.3 RESULTS COMPARISON -----
/COM,
/COM,          |  TARGET  |  ANSYS  |  RATIO
/COM,
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,'    ',F10.5,'    ',F10.5,'    ',1F5.3)
/COM,-----
/OUT          ! 结束数据重定向,关闭输出文件
FINISH        ! 退出后处理模块
*LIST, EX7_3, out ! 列表显示文件EX7_3.out的内容

```

4. 结果比较和讨论

通过以上程序计算,得到如表7.3所示的计算结果及其与理论结果的比较,可见ANSYS计算结果和理论结果完全一致,而且只利用了一个弹簧单元和集中质量单元,说明在计算考虑刚体转动惯量的扭转振动计算中,ANSYS的计算模型和结果具有非常高的计算精度。

表7.3 用弹簧单元和集中质量单元模拟圆盘的扭转自由振动频率

	理论解	Ansys解	比值
第1阶频率f1(Hz)	0.63346	0.63346	1.000

7.4 圆杆连接圆盘的扭转振动

有绕质心转动惯量 $J=0.0035026\text{kg}\cdot\text{m}^2$ 的马达固定在质量很轻的长度 $l=0.2032\text{m}$,直径为 $d=9.5248\text{mm}$ 的实心轴的一端,轴的另一端固定(如图7.12所示),试利用直管单元PIPE16和集中质量单元MASS21确定该马达在制动后自由振动的频率。忽略杆件质量对频率的影响,取剪切弹性模量 $G=82.742\text{GPa}$,弹性模量 $E=215.13\text{GPa}$ 。




图7.12 用圆截面杆连接的扭转振动圆盘模型

1. 理论解

按照文献[9]第10页练习1.3的条件和公式。忽略轴的质量影响，轴对圆盘的振动只提供刚度影响，所以先计算轴的扭转刚度。根据圆截面轴扭转的变形公式，在扭矩 M_n 作用下，长度为 l 的圆界面杆，扭转角计算公式为：

$$\phi = \frac{M_n l}{G I_p}$$

其中 $G I_p$ 是圆截面杆的扭转刚度， $I_p = \pi d^4 / 32$ 。由于扭转刚度定义为产生单位弧度所需要的扭矩，所以：

$$k = M_n / \phi = G I_p / l = 82.74210^9 \times 3.14159 \times 9.5248^4 \times 10^{-12} / 0.2032 = 328.8 \text{ N} \cdot \text{m/rad}$$

根据前面的圆盘扭转振动的频率公式：

$$f_1 = \frac{1}{2\pi} \sqrt{\frac{k}{J}} = \frac{1}{2\pi} \sqrt{\frac{G I_p}{J l}}$$

代入前面数据，得到：

$$f_1 = \frac{1}{2\pi} \sqrt{\frac{k}{J}} = \frac{1}{2 \times 3.14159} \sqrt{\frac{328.8}{0.0035026}} = 48.763 \text{ Hz}$$

2. ANSYS求解

当忽略杆件的质量对扭转振动的影响时，可以将该振动系统看作是单自由度的扭转振动系统。利用ANSYS提供的扭转刚度和质量单元来模拟，在集中质量位置设置一个旋转主自由度，实心杆的壁厚可以取外直径的一半，节点和单元布置如图7.13所示。完成该问题的模态分析，使用以下主要步骤。



图7.13 利用杆单元和质量单元模拟圆盘的扭转振动

(1) 定义工程选项、分析类型、单元类型、实常数和材料参数。

首先用“/PREP7”加载前处理模块，用“/TITLE”定义标题。用分析类型定义命令

“ANTYPE, MODAL”定义分析类型为模态(MODAL)分析,并用“MODOPT, REDUC, 1,,1”定义使用降阶方法求解,显示第1阶模态,用“MXPAND,1”命令设置只展开第1阶模态。

用单元类型定义命令“ET,1,PIPE16”定义第1类单元为直管单元PIPE16。用“ET,2, MASS21”定义第2类单元为2D质量单元MASS21。定义第1类单元的实常数“R,1, 9.5248e-3, 4.7624e-3”,表示外直径9.5248mm和壁厚4.7624mm。第2类集中质量单元的实常数“R,2,,, 3.5026E-3”,表示其转动惯量为 $3.5026\text{E-}3\text{kg}\cdot\text{m}^2$ 。

用材料参数定义命令“MP,EX,1, 215.13E9”定义第1类材料的弹性模量 $EX = 215.13\text{ GPa}$ 。用“MP,PRXY,,0.3”定义第1类材料的泊松比为0.3,省略的参数为1。

(2) 定义模型节点和单元连接关系。

用节点定义命令“N,1”在(0,0,0)位置定义结点1,用“N,2,0.2032”在(0.2032m, 0)位置定义第2个节点。用单元定义命令“E,1,2”过节点1,2定义第1个直管单元。用“TYPE,2”将默认的单元类型从直管单元,改变为第2类单元(质量单元)。用“REAL,2”将默认的实常数从弹簧刚度,改变为第2类实常数(集中质量)。命令“E,2”表示用第2个节点定义第2个单元,用主自由度定义命令“M,2,ROTX”指定2号节点的主自由度为绕X轴的转动自由度ROTX。经过这两步建立的振动模型如图7.14所示。




图7.14 用杆单元连接的圆盘振动系统的有限元模型

(3) 指定输出选项,位移约束。

用输出选项命令“OUTPR,BASIC,1”设置只输出第1阶模态的基本内容。用位移约束定义命令“D,ALL,ALL”约束所有节点的所有位移自由度。用删除位移约束命令“DDELE,2,ROTX”删除2号节点的ROTX自由度。

(4) 进入求解模块求解,开始求解。

(5) 提取计算结果。

用提取计算结果命令“*GET,FREQ,MODE,1,FREQ”提取第1阶模态的频率计算结果,存入变量FREQ。提取出的频率计算结果如图7.15所示。

(6) 定义数组变量,将计算结果输出到文本文件。



图7.15 用杆单元连接的圆盘振动的频率计算结果

3. ANSYS源程序

```

FINISH
/CLEAR, NOSTART      ! 清除数据,不加载初始化文件
!(1) 定义工程选项、分析类型、单元类型、实常数和材料参数
/FileName, EX7.4      ! 定义工程文件名称
/PREP7               ! 加载前处理模块
/TITLE, EX7.4, NATURAL FREQUENCY OF A MOTOR-GENERATOR
C*** VIBRATION THEORY AND APPLICATIONS, THOMSON, 2ND, PAGE 10, EX 1.3-3
ANTYPE,MODAL          ! 定义分析类型为模态(MODAL)分析
MODEOPT,REDUC,1,,,1   ! 用降阶方法求解,显示第1阶振型
MXPAND,1              ! 展开第1阶频率
ET,1,PIPE16           ! 定义第1类单元为直管单元PIPE16
ET,2,MASS21           ! 定义第2类单元为2D质量单元MASS21
R,1,9.5248e-3,4.7624e-3 ! 定义管单元实常数,外直径9.5248mm和壁厚4.7624mm
R,2,,,,3.5026E-3       ! 定义第2类实常数,转动惯量3.5026E-3kg.m^2
MP,EX,1,215.13E9       ! 定义第1类材料的弹性模量EX=215.13 GPa
MP,PRXY,,0.3           ! 定义第1类材料的泊松比为PRXY=0.3
!(2) 定义模型节点和单元连接关系
N,1                   ! 在(0,0,0)位置定义结点1
N,2,0.2032            ! 在(0.2032m,0,0)位置定义第2个节点
E,1,2                 ! 过节点1,2定义直管单元
TYPE,2                ! 将默认的单元类型从弹簧单元,改变为第2类单元(质量单元)
REAL,2                ! 将默认的实常数从弹簧刚度,改变为第2类实常数(集中质量)
E,2                   ! 用第2个节点定义第2个单元
M,2,ROTX              ! 指定2号节点的主自由度为绕x轴的转动自由度ROTX
!(3) 指定输出选项,位移约束
OUTPR,BASIC,1         ! 输出第1阶模态的基本内容(包括节点基本结果,节点反力和单元解)
D,ALL,ALL             ! 约束所有节点的所有位移自由度
DDELE,2,ROTX          ! 放松2号节点的ROTX自由度

```

```

FINISH      ! 退出前处理模块PREP7
!(4) 进入求解模块求解
/SOLU       ! 加载求解模块
SOLVE       ! 开始进行模态分析的求解
!(5) 提取计算结果
*GET,FREQ,MODE,1,FREQ      ! 提取第1阶模态的频率计算结果,存入变量FREQ
!(6) 定义数组变量,将计算结果输出到文本文件
*DIM,LABEL,CHAR,1,2
*DIM,VALUE,,1,3
LABEL(1,1) = '      F,'
LABEL(1,2) = '(Hz)  '
*VFILL,VALUE(1,1),DATA,48.763
*VFILL,VALUE(1,2),DATA,FREQ
*VFILL,VALUE(1,3),DATA,ABS(FREQ/48.763)
/COM
/OUT, EX7_4, out      ! 将输出内容重定向到文件EX7_4.out
/COM,----- EX7.4 RESULTS COMPARISON -----
/COM,
/COM,                |  TARGET  |  ANSYS  |  RATIO
/COM,
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,'      ',F10.3,'      ',F10.3,'      ',1F5.3)
/COM,-----
/OUT      ! 结束数据重定向,关闭输出文件
FINISH    ! 退出后处理模块
*LIST, EX7_4, out      ! 列表显示文件EX7_4.out的内容

```

4. 结果比较和讨论

通过以上程序计算,得到如表7.4所示的计算结果及其与理论结果的比较。虽然计算模型只用了一个弹簧单元和集中质量单元,但是ANSYS计算结果和理论结果极其接近。说明在考虑刚体转动惯量的扭转振动计算中,利用考虑扭转刚度的直管单元和集中质量组成的振动系统,ANSYS的计算结果具有非常高的计算精度。

表7.4 用直管单元和集中质量单元模拟圆盘的扭转自由振动频率

	理论解	Ansysis解	比值
第1阶频率f1(Hz)	48.763	48.780	1.000

7.5 钻杆的扭转自由振动

长度 $l=1524\text{m}$ 的石油钻杆,上端固定,下端部有长度可以忽略的集中质量(如图7.16所示),试利用直管单元PIPE16,梁单元BEAM4和集中质量单元MASS21,确定该钻杆自

由振动的前两阶频率。已知钻杆的外直径 $OD=114.3\text{mm}$ ，内直径 $ID=97.28\text{mm}$ ，横截面的极惯性矩 $I_p=7.962\times 10^{-6}\text{m}^4$ ，端部集中质量的转动惯量 $J_0=39.73\text{kg}\cdot\text{m}^2$ ，密度 $\rho=7855.3\text{kg/m}^3$ ，剪切弹性模量 $G=82.742\text{GPa}$ ，泊松比 $\mu=0.3$ ，弹性模量 $E=215.134\text{GPa}$ 。

1. 理论解

钻杆的转动惯量 $J_{\text{rod}}=I_p \quad l/g=96.8\text{kg}\cdot\text{m}^2$ ，

根据文献[9]，频率方程为：

$$\tan \beta l = J_{\text{rod}}/J_0=2.44$$

查表，可以得到 $\beta=1.135, 3.722, \dots$

$$= l \sqrt{\frac{\rho}{Gg}}=0.470$$

由此解出前两阶频率为：

$$f_1=1.135/0.470=2.41\text{rad./sec.}=0.384\text{Hz}$$

$$f_2=3.722/0.470=7.93\text{rad./sec.}=1.26\text{Hz}$$

2. ANSYS计算

该钻杆首先使用直管单元PIPE16进行建模，并使用子空间迭代方法计算了自由振动的频率，随后又使用梁单元BEAM4建模求解。这里使用的弹性模量可以用换算公式 $E=2G(1+\mu)$ ，壁厚可以用外直径减去内直径的一半，即 $(OD-ID)/2$ 来计算。

在使用梁单元BEAM4进行计算时，极惯性矩的输入可以省略，因为它始终等于两个抗弯曲惯性矩之和，即 $I_{YY}+I_{ZZ}$ ，而这两个惯性矩是输入的基本截面几何参数。

这里选择了12个扭转主自由度，在钻头位置有一个，其余的11个都布置在管单元上（如图7.17所示）。以下是完成这些计算的主要步骤。




图7.16 用圆截面钻杆和钻头扭转振动模型



图7.17 用梁单元和集中质量单元模拟钻杆振动

(1) 定义工程选项、单元类型、实常数和材料参数。

用“/FileName, EX7.5”定义文件名称，用“/PREP7”加载前处理模块。用“/TITLE”命令定义标题。

首先使用直线管单元PIPE16进行计算。用单元类型定义命令“ET,1,PIPE16”定义第1类单元为直管单元PIPE16，用“ET,2,MASS21”定义第2类单元为2D质量单元MASS21。

用实常数定义命令“R,1,0.1143,0.0085”定义第1类实常数：管单元的外直径114.3mm和壁厚8.5mm。用“R,2,,,,,39.73”定义第2类实常数：集中质量的转动惯量 $39.73 \text{ kg}\cdot\text{m}^2$ 。

用材料参数定义命令“MP,EX,1,215.134E9”定义第1类材料的弹性模量 $EX=215.134 \text{ GPa}$ 。用“MP,DENS,1,7855.3”定义第1类材料的密度 $DENS=7855.3 \text{ kg/m}^3$ ，用“MP,NUXY,1,0.3”定义第1类材料的泊松比 $NUXY=0.3$ 。

(2) 定义模型节点和单元连接关系。

为了便于观察模型建立过程，用视点定义命令“/VIEW,1,1,1,1”将视点切换到(1,1,1)等轴测位置。

用节点位置定义命令“N,1”在坐标原点(0,0,0)定义第1个结点，用“N,13,,,-1524”在(0,0,-1524m)位置定义第2个节点。用在两个节点之间填充生成节点命令“FILL”在1号和13号节点中间填充生成其余节点。

用单元定义命令“E,1,2”过节点1,2定义第一个管单元。用单元生成命令“EGEN,12,1,1”循环12次，按照前面定义的单元模式，每个单元的两端节点号增加1，生成其余的11个单元。

用当前单元类型设置命令“TYPE,2”将默认的单元类型从第1类弹簧单元，改变为第2类单元（质量单元）。用“REAL,2”将默认的实常数从第1类弹簧刚度，改变为第2类实常数（集中质量）。用单元定义命令“E,13”在第13号节点处定义单元。

(3) 进入求解模块，指定输出选项和位移约束。

用“/SOLU”加载求解模块/SOLU。用分析类型定义命令“ANTYPE,MODAL”定义分析类型为模态分析，并用“MODOPT,SUBSP,2”指定用子空间迭代法计算前2阶模态。

用位移约束定义命令“D,1,UX,,,13,,UY,UZ,ROTX,ROTY”约束从1到13号节点的除了ROTZ之外的所有自由度，即UX,UY,UZ,ROTX和ROTY。用“D,1,ROTZ”约束第1号节点的ROTZ自由度。用这些步骤完成的有限元模型如图7.18所示。




图7.18 用杆单元连接的圆盘振动系统的有限元模型

(4) 设置输出第1阶频率的默认信息，并开始求解。

用输出结果设置命令“OUTPR,,1”定义只输出第1阶频率的基本内容,包括节点基本结果,节点反力和单元解。这里的空缺项与“BASIC”选项相同。用求解命令“SOLVE”开始进行模态分析的求解。

(5) 进入后处理模块,提取前两阶频率的计算结果。

用“/POST1”命令进入一般后处理模块,用结果提取命令“*GET,FREQ1,MODE,1,FREQ”提取第1阶模态的频率计算结果,存入变量FREQ1;用“*GET,FREQ2,MODE,2,FREQ”提取第2阶模态的频率计算结果,存入变量FREQ2。利用该命令提取出的计算结果如图7.19所示。

(6) 定义数组变量,将第一步的计算结果保存到内存变量。

用“*DIM,LABEL,CHAR,2,2”定义2*2的字符型数组LABEL,用“*DIM,VALUE,,2,3”定义2*3的数值型数组VALUE。利用等号“=”和“*VFILL”命令给这些变量赋值理论结果和有限元计算结果。

并利用命令“SAVE, TABLE_1”将这组结果数据保存到默认数据库TABLE_1。

(7) 重新进入前处理模块,将单元类型改变为三维梁单元BEAM4,重新分析该问题。

用单元类型定义命令“ET,1,BEAM4”定义第1类单元为三维梁单元BEAM4。用实常数定义命令“R,1,0.0028283,3.982E-6,3.982E-6,0.1143,0.1143”定义三维梁单元BEAM4的截面积为 8.283 cm^2 ,两个方向的抗弯惯性矩为 $9.82\text{E-}6 \text{ m}^4$,两个方向的截面高度为 0.1143 m ,并利用“RMORE,,7.962E-6”定义梁单元的扭转惯性矩为 $IX=7.962\text{e-}6 \text{ m}^4$ 。

(8) 进入求解模块,指定模态分析的选项,并开始求解。

用“/SOLU”加载求解模块,用“MODOPT,REDUC,2,,,2”定义用降阶方法求解前两阶模态。用主自由度定义命令“M,2,ROTZ,13”指定2号节点到13号节点的绕Z轴转动自由度为主自由度。用“SOLVE”命令开始求解前两阶模态。

(9) 进入一般后处理模块,提取频率计算结果。

用“/POST1”命令进入一般后处理模块,用“*GET,FREQ1,MODE,1,FREQ”提取第1阶模态的频率计算结果,存入变量FREQ1,用“*GET,FREQ2,MODE,2,FREQ”提取第2阶模态的频率计算结果,存入变量FREQ2。提取出的频率计算结果如图7.20所示。



图7.19 用直管单元连接的圆盘振动的频率计算结果 图7.20 用梁单元连接的圆盘振动的频率计算结果

(10) 定义数组变量，将计算结果输出到文本文件。

3. ANSYS源程序

```

FINISH
/CLEAR, NOSTART          ! 清除数据,不加载初始化文件
!(1) 定义工程选项、单元类型、实常数和材料参数
/FileName, EX7.5          ! 定义文件名称
/PREP7                   ! 加载前处理模块
/TITLE, EX7.5, TORSIONAL FREQUENCIES OF A DRILL PIPE
C*** VIBRATION THEORY AND APPLICATIONS, THOMSON, 2ND., PAGE 272, EX.8.4-5
C*** 使用直线管单元PIPE16计算
ET,1,PIPE16              ! 定义第1类单元为直管单元PIPE16
ET,2,MASS21              ! 定义第2类单元为2D质量单元MASS21
R,1,0.1143,0.0085        ! 定义第1类实常数:管单元的外直径114.3mm和壁厚8.5mm
R,2,,,,, 39.73           ! 定义第2类实常数,集中质量的转动惯量39.73 kg.m^2
MP,EX,1,215.134E9        ! 定义第1类材料的弹性模量EX=215.134 GPa
MP,DENS,1,7855.3         ! 定义第1类材料的密度DENS=7855.3 kg/m^3
MP,NUXY,1,0.3            ! 定义第1类材料的泊松比NUXY=0.3
!(2) 定义模型节点和单元连接关系
/VIEW,1,1,1,1           ! 将视点切换到(1,1,1)等轴测位置
N,1                      ! 在坐标原点(0,0,0)位置定义第1号结点
N,13,,,,-1524           ! 在(0,0,-1524m)位置定义第2个节点
FILL                     ! 在1号和13号节点中间填充生成其余节点
E,1,2                   ! 过节点1,2定义管单元
EGEN,12,1,1             ! 按照前面定义的单元模式,循环12次,生成其余的11个单元
TYPE,2                  ! 将默认的单元类型从弹簧单元,改变为第2类单元(质量单元)
REAL,2                  ! 将默认的实常数从弹簧刚度,改变为第2类实常数(集中质量)
E,13                    ! 用第13个节点定义单元
FINISH                  ! 退出前处理模块PREP7
!(3) 进入求解模块,指定输出选项和位移约束
/SOLU                   ! 加载求解模块
ANTYPE,MODAL            ! 定义分析类型为模态分析
MODOPT,SUBSP,2          ! 指定用降阶方法求前2阶模态
D,1,UX,,,13,,,UY,UZ,ROTX,ROTY ! 约束从1到13号节点的除了ROTX,ROTY之外的所有自由度
D,1,ROTX                ! 约束1号节点的ROTX自由度
!(4) 设置输出第1阶频率信息,并开始求解
OUTPR,,1               ! 输出第1阶频率的基本内容(包括节点基本结果,节点反力和单元解)
SOLVE                  ! 开始求解前2阶模态
FINISH                 ! 退出求解模块SOLU
!(5) 进入后处理模块,提取前两阶频率的计算结果
/POST1                 ! 进入一般后处理模块
*GET,FREQ1,MODE,1,FREQ  ! 提取第1阶模态的频率计算结果,存入变量FREQ1
*GET,FREQ2,MODE,2,FREQ  ! 提取第2阶模态的频率计算结果,存入变量FREQ2

```

```

!(6) 定义数组变量,将计算结果输出到文本文件
*DIM,LABEL,CHAR,2,2      ! 定义2*2的字符型数组LABEL
*DIM,VALUE,,2,3          ! 定义2*3的数值型数组VALUE
LABEL(1,1) = '      f1','      f2'
LABEL(1,2) = ', Hz      ','      Hz      '
*VFILL,VALUE(1,1),DATA,0.384,1.26
*VFILL,VALUE(1,2),DATA,FREQ1,FREQ2
*VFILL,VALUE(1,3),DATA,ABS(FREQ1/0.384),ABS(FREQ2/1.26)
SAVE, TABLE_1          ! 保存数据到默认数据库TABLE_1
FINISH                  ! 退出后处理模块
!(7) 重新进入前处理模块,将单元类型改变为BEAM4,重新分析该振动问题
/PREP7                  ! 加载前处理模块
C****                  使用梁单元BEAM4
ET,1,BEAM4              ! 定义第1类单元为三维梁单元BEAM4
R,1,0.0028283,3.982E-6,3.982E-6,0.1143, 0.1143      ! 定义三维梁单元的截面几何参数:截
面积为28.283cm^2,两个抗弯惯性矩为3.982E-6 m^4,两个方向的截面高度为0.1143m.
RMORE,,7.962E-6          ! 还有梁单元的扭转惯性矩为IIX=7.962e-6 m^4.
FINISH                  ! 退出前处理模块PREP7
!(8) 进入求解模块,指定模态分析的选项,并开始求解
/SOLU                  ! 加载求解模块
MODEPT,REDUC,2,,,2      ! 指定用降阶方法计算前两阶模态
M,2,ROTZ,13             ! 指定2号节点到13号节点的绕z轴转动自由度ROTZ为主自由度
SOLVE                  ! 开始模态分析的求解
FINISH                  ! 退出求解模块SOLUTION
!(9) 进入一般后处理模块,提取频率计算结果
/POST1                 ! 进入一般后处理模块
*GET,FREQ1,MODE,1,FREQ   ! 提取第1阶模态的频率计算结果,存入变量FREQ1
*GET,FREQ2,MODE,2,FREQ   ! 提取第2阶模态的频率计算结果,存入变量FREQ2
!(10) 定义数组变量,将计算结果输出到文本文件
LABEL(1,1) = 'UR(INNER'
LABEL(1,2) = ' RAD),in'
*VFILL,VALUE(1,1),DATA,.384,1.26
*VFILL,VALUE(1,2),DATA,FREQ1,FREQ2
*VFILL,VALUE(1,3),DATA,ABS(FREQ1/.384),ABS(FREQ2/1.26)
SAVE, TABLE_2          ! 保存数据到数据库TABLE_2
RESUME, TABLE_1
/COM
/OUT, EX7_5, out         ! 将输出内容重定向到文件EX7_5.out
/COM,----- EX7.5 RESULTS COMPARISON -----
/COM,
/COM,                   |  TARGET   |  ANSYS   |  RATIO
/COM,
/COM,RESULTS USING PIPE16 ELEMENTS:
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)

```

```

(1X,A8,A8,' ',F10.4,' ',F10.4,' ',1F5.3)
/NOPR
RESUME, TABLE_2
/GOPR
/COM,
/COM, RESULTS USING BEAM4 ELEMENTS:
*VWRITE, LABEL(1,1), LABEL(1,2), VALUE(1,1), VALUE(1,2), VALUE(1,3)
(1X,A8,A8,' ',F10.4,' ',F10.4,' ',1F5.3)
/COM,-----
/OUT          ! 结束数据重定向, 关闭输出文件
FINISH        ! 退出后处理模块
*LIST, EX7_5, out      ! 列表显示文件EX7_5.out的内容

```

4. 结果比较和讨论

通过以上程序两种不同单元的计算, 得到如表7.5所示的计算结果及其与理论结果的比较, 可见ANSYS计算结果和理论结果略有差异。但是, 直管单元PIPE16和梁单元BEAM4之间完全一致。而且只利用了十几个单元, 说明在考虑连续变形体的转动惯量的扭转振动计算中, ANSYS的计算模型和结果仍然具有非常高的计算精度。如果这里的节点数量和单元增加一些, 计算结果继续接近理论值, 读者可以验证一下。

表7.5 用直管单元, 梁单元和集中质量单元模拟圆盘的扭转自由振动频率

直管单元PIPE16	理论解	Ansys解	比值
第1阶频率f1 (Hz)	0.3840	0.3832	0.998
第2阶频率f2 (Hz)	1.2600	1.2630	1.002
三维梁单元BEAM4	理论解	Ansys解	比值
第1阶频率f1 (Hz)	0.3840	0.3832	0.998
第2阶频率f2 (Hz)	1.2600	1.2630	1.002